"Mariel" by Michael Baltes

Include German by Team Gerx.
Include German Default Messages by Team Gerx.
Include Flexible Windows by Jon Ingold.
Include Glimmr Graphic Hyperlinks by Erik Temple.
Include Book Interface by Michael Baltes.

Part - The prelude

Section - Release Options

Release along with cover art.

Section - Additional pictures

Figure police is the file "police.jpg".
Figure leftpath is the file "leftpath.jpg".
Figure rightpath is the file "rightpath.jpg".
Figure dyingforest is the file "dyingforest.jpg".
Figure backyard is the file "backyard.jpg".
Figure dumping is the file "dumping.jpg".
Figure street is the file "street.jpg".
Figure clearing is the file "clearing.jpg".
Figure shaft is the file "shaft.jpg"

Section Custom Library Messages

Table of custom library messages (continued)
library-action	library-message-id	library-message-text
--	1	"(Es werden nur die ersten sechzehn Objekte berücksichtigt.)[ignore library line break]"
--	2	"Hier gibt es nichts zu tun!"
--	3	" Ich bin gestorben [ignore library line break]"
--	4	" Ich habe gewonnen [ignore library line break]"
--	5	"Soll ich [ignore library line break]"
--	6	"[bracket]Der Interpreter kann leider keine Spielzüge rückgängig machen.[close bracket]"
--	7	"[#if Z-machine]Der Spielzug konnte nicht rückgängig gemacht werden. [bracket]Nicht alle Interpreter verfügen über diese Funktionalität.[close bracket][#otherwise][bracket]Ich kann keine weiteren Spielzüge rückgängig machen.[close bracket][#end if]"
--	8	"Bitte antworte mit einer der oben genannten Möglichkeiten."
--	9	"Es ist nun stockfinster hier!"
--	10	"Wie bitte?"
--	11	"[bracket]Erst etwas tun, dann rückgängig machen.[close bracket]"
--	12	"[bracket]Ich kann leider immer nur einen Zug in Folge rückgängig machen.[close bracket]"
--	13	"[bracket]Der letzte Zug wurde rückgängig gemacht.[close bracket]"
--	14	"Da gibt es leider nichts zu korrigieren."
--	15	"Denk [dir] nichts weiter."
--	16	"Mit 'hoppla' kann immer nur ein Wort korrigiert werden."
--	17	"Es ist stockfinster, ich sehe nichts."
--	18	"[player]"
--	19	"Gutaussehend wie immer."
--	20	"Wenn [du] einen Befehl wie ~Häschen, hüpf~ wiederholen willst, sag 'nochmal', nicht 'Häschen, nochmal'."
--	21	"Das kann man nicht wiederholen."
--	22	"Ein Satz kann nicht mit einem Komma beginnen."
--	23	"Ich soll wahrscheinlich jemandem eine Anweisung erteilen, aber mir ist nicht klar wem."
--	24	"Mit [dem library message object] kann man nicht reden."
--	25	"Um mit jemandem zu reden, benutze bitte 'Jemand, hallo'."
--	26	"(Dazu hebst [du] [den noun] erst auf.)[ignore library line break]"
--	27	"[explicit error message]Diesen Satz habe ich nicht verstanden."
--	28	"[#if explicit error messages is active][explicit error message]Diesen Satz habe ich nicht verstanden.[#otherwise]Ich habe nur Folgendes verstanden: [ignore library line break][#end if]"
--	29	"Diese Zahl habe ich nicht verstanden."
--	30	"[explicit error message]So etwas kannst [du] hier nicht sehen."
--	31	"Du scheinst nicht alles gesagt zu haben!"
--	32	"Aber das hast [du] nicht bei [dir]!"
--	33	"Hier kannst [du] nur ein Objekt angeben."
--	34	"Du kannst in jedem Satz nur einmal Listen von Objekten angeben."
--	35	"Mir ist nicht klar, worauf sich '[pronoun dictionary word]' bezieht."
--	36	"Du hast etwas ausgeschlossen, das gar nicht zur Ausgangsmenge gehört!"
--	37	"Das kannst [du] nur mit Lebewesen sinnvoll machen."
--	38	"[explicit error message]Ich habe dieses Verb nicht verstanden."
--	39	"Damit brauchst [du] [dich] in diesem Spiel nicht zu beschäftigen."
--	40	"Die siehst '[pronoun dictionary word]' ([pronoun reference object]) hier im Moment nicht."
--	41	"Das Satzende habe ich leider nicht verstanden."
--	42	"[if library message amount is 0]Nichts davon steht[end if][if library message amount is 1]Nur eines davon steht[end if][if library message amount > 1]Nur [library message amount in words] davon stehen[end if] zur Verfügung."
--	43	"Es gibt hier nichts zu tun!"
--	44	"Hier ist aber nichts, womit du das tun kannst."
--	45	"Wen meinst [du], [ignore library line break]"
--	46	"Was meinst [du], [ignore library line break]"
--	47	"Hier kannst [du] nur ein Objekt angeben. Welches?"
--	48	"[Wem-oder-Wen-Frage]"
--	49	"[Womit-oder-Was-Frage]"
--	50	"Du hast gerade [if library message amount > 0][library message amount][otherwise][0 - library message amount][end if] Punkt[if library message amount is not 1 and library message amount is not -1]e[end if] [if library message amount <= 0]verloren[otherwise]bekommen[end if]."
--	51	"(Da etwas Dramatisches passiert ist, wurde die Liste [deiner] Anweisungen nicht komplett ausgeführt.)"
--	52	"[line break]Bitte eine Nummer von 1 bis  [library message amount] angeben, mit 0 die Anzeige auffrischen oder die EINGABETASTE drücken."
--	53	"[line break][bracket]Bitte die LEERTASTE drücken.[close bracket]"
--	54	"[bracket]Kommentar notiert.[close bracket]"
--	55	"[bracket]Kommentar NICHT notiert.[close bracket]"
--	56	"."
--	57	"?"
--	58	"[The person asked] [if person asked is plural-named]können[otherwise]kann[end if] das nicht tun."
--	59	"Du mu[ss]t ein Hauptwort angeben."
--	60	"Du kannst kein Hauptwort angeben."
--	61	"Du mu[ss]t den Namen eines Objekts angeben."
--	62	"Hier kannst [du] keinen Namen eines Objekts angeben."
--	63	"Du mu[ss]t noch ein zweites Objekt angeben."
--	64	"Hier kannst [du] kein zweites Objekt angeben."
--	65	"Du mu[ss]t ein zweites Hauptwort angeben."
--	66	"Du kannst hier kein zweites Hauptwort angeben."
--	67	"Versuch es mit etwas mit mehr Substanz."
--	68	"([The person asked] nimmt zuerst [den noun].)"
--	69	"(Dazu hebst [du] [den noun] erst auf.)"
--	70	"Die Verwendung von UNDO ist in diesem Spiel nicht erlaubt."
--	71	"Dunkelheit"
--	72	"[Der noun] [hat] Besseres zu tun."
--	73	"Dieses Hauptwort macht in diesem Zusammenhang keinen Sinn."
--	74	"[bracket][The person asked] kann keine ans Spiel gerichteten Anweisungen ausführen.[close bracket]"
--	75	"  Ende  [ignore library line break]"
--	76	" oder "
--	98	"Ich verstehe das Wort '[original word]' in diesem Zusammenhang nicht"
--	99	"Ich kenne das Wort '[original word]' nicht."
--	101	" (Licht spendend)"
--	102	" ([*der* noun] geschlossen [ist])"
--	104	" ([*der* noun] leer [ist])"
--	106	" ([*der* noun] geschlossen und leer [ist])"
--	103	" (geschlossen und Licht spendend)"
--	105	" (leer und Licht spendend)"
--	107	" (geschlossen, leer and Licht spendend)"
--	108	" (Licht spendend und angezogen"
--	109	" (Licht spendend"
--	110	" (angezogen"
--	111	" ([*der* noun]"
--	112	"offen [ist noun]"
--	113	"offen, aber leer [ist noun]"
--	114	"geschlossen [ist noun]"
--	115	"abgeschlossen [ist noun]"
--	116	" und leer [ist noun]"
--	117	" ([*der* noun] leer [ist])"
--	118	" [er noun] [if noun is plural-named]enthalten[otherwise]enthält[end if] "
--	119	" (darauf "
--	120	", daauf "
--	121	" (darin "
--	122	", darin "
Answering it that action	1	"Keine Antwort."
Asking it about action	1	"Keine Antwort."
Asking it for action	1	"Keine Antwort."
Attacking action	1	"Gewalt ist keine Lösung."
Burning action	1	"Das ist gefährlich und wenig sinnvoll."
Buying action	1	"Hier gibt es nichts zu kaufen."
Climbing action	1	"Damit wirst [du] nichts erreichen."
Closing action	1	"Du kannst [den noun] nicht schlie[ß]en."
Closing action	2	"[Der noun] ist bereits geschlossen."
Closing action	3	"Du schlie[ß]t [den noun]."
Closing action	4	"[Der person asked] schlie[ß][t] [den noun]."
Closing action	5	"[Der noun] schlie[ß][t] sich."
Consulting it about action	1	"In [dem noun] findest [du] nichts Interessantes darüber."
Consulting it about action	2	"[Der person asked] sieht sich [den noun] an."
Cutting action	1	"Es bringt nichts, [den noun] zu zerschneiden."
Drinking action	1	"Das kann man nicht trinken."
Dropping action	1	"[Der noun] [ist] bereits hier."
Dropping action	2	"Du hast [den noun] gar nicht."
Dropping action	3	"(Dazu ziehst [du] [den noun] erst aus.)[ignore library line break]"
Dropping action	4	"In Ordnung."
Dropping action	5	"Es ist kein Platz mehr auf [dem noun]."
Dropping action	6	"Es ist kein Platz mehr in [dem noun]."
Dropping action	7	"[Der person asked] legt [den noun] hin."
Eating action	1	"[Der noun] ist nicht e[ss]bar."
Eating action	2	"Du i[ss]t [den noun]. Nicht schlecht."
Eating action	3	"[Der person asked] i[ss]t [den noun]."
Entering action	1	"Aber du bist doch schon [if noun is a supporter]auf[otherwise]in[end if] [dem noun]."
Entering action	2	"Du kannst auf [dem noun] nicht [quoted verb posture]."
Entering action	3	"Du kannst nicht in [den noun] wenn [er] geschlossen [ist]."
Entering action	4	"Du kannst nur frei stehende Dinge betreten."
Entering action	5	"Du [first person quoted verb posture] [if noun is a supporter]auf[otherwise]in[end if] [den noun]."
Entering action	6	"(Du steigst zuerst [if the library message object is a supporter]von[otherwise]aus[end if] [dem library message object].)[ignore library line break]"
Entering action	7	"(Du steigst zuerst [if the library message object is a supporter]auf[otherwise]in[end if] [den library message object].)[ignore library line break]"
Entering action	8	"[Der person asked] betritt [den noun]."
Entering action	9	"[Der person asked] steigt auf [den noun]."
Examining action	1	"Es ist dunkel hier."
Examining action	2	"Du siehst nichts Besonderes an [dem noun]."
Examining action	3	"[Der noun] [ist] im Moment [if noun is switched on]an[otherwise]aus[end if]."
Examining action	4	"[Der person asked] betrachte[t] [den noun] genau."
Examining action	5	"Du siehst nichts Unerwartetes in dieser Richtung."
Exiting action	1	"Aber [du] bist im Moment nirgendwo drin."
Exiting action	2	"Du kannst [den library message object] nicht verlassen, wenn [er] geschlossen [ist]."
Exiting action	3	"You steigst [if library message object is a supporter]von[otherwise]aus[end if] [dem library message object]."
Exiting action	4	"Aber [du] bist gar nicht [if library message object is a supporter]auf[otherwise]in[end if] [dem library message object]."
Exiting action	5	"[Der person asked] steig[t] von [dem library message object]."
Exiting action	6	"[Der person asked] steig[t] aus [dem library message object]."
Getting off action	1	"Aber [du] bist gar nicht auf [dem library message object]."
Giving it to action	1	"Du hast [den noun] gar nicht."
Giving it to action	2	"Jetzt willst [du] es [dir] aber mal richtig geben, was?"
Giving it to action	3	"[Der second noun] schein[t] nicht besonders interessiert zu sein."
Giving it to action	4	"[Der second noun] [ist] nicht in der Lage, etwas anzunehmen."
Giving it to action	5	"Du gibst [dem noun] [den second noun]."
Giving it to action	6	"[Der person asked] [if person asked is plural-named]geben[otherwise]gibt[end if] [dir] [den noun]."
Giving it to action	7	"[Der person asked] [if person asked is plural-named]geben[otherwise]gibt[end if] [dem second noun] [den noun]."
Going action	1	"Das geht nicht, solange [du] noch [if library message object is a supporter]auf[otherwise]in[end if] [dem library message object] bist."
Going action	2	"Du kannst nicht in diese Richtung gehen."
Going action	3	"Es gibt hier keine offensichtlichen Ausgänge."
Going action	4	"(nach [noun])"
Going action	5	"Es gibt mehrere Ausgänge. Bitte sage genau, wohin [du] gehen möchtest."
Going action	6	"Das geht nicht, [der library message object] führ[t] nirgendwohin."
Going action	7	"Du mu[ss]t sagen, in welche Richtung [du] gehen möchtest."
Going action	8	"[Der person asked] geh[t] hinauf"
Going action	9	"[Der person asked] geh[t] hinunter"
Going action	10	"[Der person asked] geh[t] nach [noun]"
Going action	11	"[Der person asked] kommt[t] von oben"
Going action	12	"[Der person asked] kommt[t] von unten"
Going action	13	"[Der person asked] kommt[t] von [library message object]"
Going action	14	"[Der person asked] kommt[t] herein"
Going action	15	"[Der person asked] erreich[t] [den library message object] von oben"
Going action	16	"[Der person asked] erreich[t] [den library message object] von unten"
Going action	17	"[Der person asked] erreich[t] [den library message object] von [second noun]"
Going action	18	"[Der person asked] geh[t] durch [den library message object]"
Going action	19	"[Der person asked] kommt[t] von [dem library message object]"
Going action	20	"auf [dem library message object]"
Going action	21	"in [dem library message object]"
Going action	22	", [den library message object] vor sich her schiebend, und [dich] dazu"
Going action	23	", [den library message object] vor sich her schiebend"
Going action	24	", [den library message object] wegschiebend"
Going action	25	", [den library message object] hereinschiebend"
Going action	26	", [dich] mitziehend"
Going action	27	"(Dazu steigst [du] erst [if library message object is a supporter]von[otherwise]aus[end if][dem library message object].)[ignore library line break]"
Going action	28	"(Du öffnest zuerst [den library message object].)[ignore library line break]"
Inserting it into action	1	"Du mu[ss]t [den noun] in der hand halten, um [ihn] in etwas anderes legen zu können."
Inserting it into action	2	"[Der noun] [ist] weder ein Behälter noch eine Ablage."
Inserting it into action	3	"[Der noun] [ist] closed."
Inserting it into action	4	"Dazu mu[ss]t [du] [den noun] erst ausziehen."
Inserting it into action	5	"Du kannst nichts in sich selbst legen."
Inserting it into action	6	"(Du ziehst [den noun] erst aus.)[ignore library line break]"
Inserting it into action	7	"Es ist kein Platz mehr in [dem noun]."
Inserting it into action	8	"In Ordnung."
Inserting it into action	9	"Du [legst] [den noun] in [den second noun]."
Inserting it into action	10	"[Der person asked] leg[t] [den noun] in [den second noun]."
Jumping action	1	"Du springst etwas motivationslos auf der Stelle."
Kissing action	1	"Konzentrier [dich] auf das Spiel."
Listening to action	1	"Du hörst nichts Unerwartetes."
Locking it with action	1	"Du kannst [den noun] nicht abschlie[ß]en."
Locking it with action	2	"[Der noun] [ist] schon abgeschlo[ss]en."
Locking it with action	3	"Du mu[ss]t [den noun] dazu erst zumachen."
Locking it with action	4	"[Der noun] pa[ss][t] nicht."
Locking it with action 5 "Du schlie[[unicode 2013]t [den noun] ab."
Locking it with action 6 "[The person asked] lock[s] [the noun]."
Looking action	1	" (auf [dem library message object])[ignore library line break]"
Looking action	2	" (in [dem library message object])[ignore library line break]"
Looking action	3	" (als [library message object])[ignore library line break]"
Looking action	4	"Auf [dem library message object] [is-are a list of things on the library message object with sublists][if the non-nested lists option is active][ignore library line break][end if]"
Looking action	7	"Dort siehst [du] nichts Au[ß]ergewöhnliches."
Looking action	8	"[if the library message object is a supporter] (auf [otherwise] (in [end if][dem library message object])[ignore library line break]"
Looking action	9	"[Der person asked] [if person asked is plural-named]sehen[otherwise]sieht sich um."
Looking under action	1	"Aber es ist dunkel."
Looking under action	2	"Du findest nichts Interessantes."
Looking under action	3	"[Der person asked] [if person asked is plural-named]sehen[otherwise]sieht unter [den noun]."
Opening action	1	"Du kannst [den noun] nicht öffnen."
Opening action	2	"[Der noun] [ist] offenbar abgeschlossen."
Opening action	3	"[Der noun] [ist] bereits offen."
Opening action	4	"Ich öffne [den noun] und finde [a list of things in the noun with accusative] darin."
Opening action	5	"Ich öffne [den noun]."
Opening action	6	"[Der person asked] öffne[t] [den noun]."
Opening action	7	"[Der noun] öffne[t] sich."
preferring abbreviated room descriptions action	1	" ist nun im knappen Modus, in dem Raumbeschreibungen nur beim ersten Betreten eines Raums angezeigt werden."
preferring sometimes abbreviated room descriptions action	1	" ist nun im ausführlichen Modus, der immer die langen Raumbeschreibungen zeigt, auch wenn [du] schon einmal dort warst."
preferring unabbreviated room descriptions action	1	" ist nun im superknappen Modus, der immer nur die kurze Raumbeschreibung anzeigt, auch wenn [du] einen Raum zum ersten Mal betrittst."
Pulling action	1	"[Der noun] [ist] unbeweglich."
Pulling action	2	"Ich bin dazu nicht in der Lage."
Pulling action	3	"Nichts passiert."
Pulling action	4	"Ich finde, das wäre sehr unhöflich."
Pulling action	5	"[Der person asked] zieh[t] [den noun]."
Pushing action	1	"[Der noun] [ist] unbeweglich."
Pushing action	2	"Ich bin dazu nicht in der Lage."
Pushing action	3	"Nichts passiert."
Pushing action	4	"Ich finde, das wäre sehr unhöflich."
Pushing action	6	"[Der person asked] schieb[t] [den noun]."
Pushing it to action	1	"[Den noun] kann ich nicht von einem Ort zum anderen schieben."
Pushing it to action	2	"Das ist keine Richtung."
Pushing it to action	3	"Nein, in diese Richtung geht das nicht."
Putting it on action	1	"Ich mu[ss] [den noun] in der Hand halten, um [ihn] auf etwas anderem ablegen zu können."
Putting it on action	2	"Ich kann nichts auf sich selbst ablegen."
Putting it on action	3	"Dinge auf [den noun] zu tun bringt vermutlich nichts."
Putting it on action	4	"Mir fehlt die nötige Geschicklichkeit."
Putting it on action	5	"(Ich ziehe [den noun] erst aus.)[ignore library line break]"
Putting it on action	6	"Es ist kein Platz mehr auf [dem noun]."
Putting it on action	7	"In Ordnung."
Putting it on action	8	"Du [legst] [den noun] auf [den second noun]."
Putting it on action	9	"[Der person asked] leg[t] [den noun] auf [den second noun]."
Quitting the game action	1	"Bitte antworte mit Ja oder Nein. "
Quitting the game action	2	"Möchtest [du] das Spiel wirklich beenden? "
Removing it from action	1	"[Der noun] [ist] leider geschlossen."
Removing it from action	2	"Aber [der noun] [ist] gar nicht [if second noun is supporter]auf[otherwise]in[end if] [dem second noun]."
Removing it from action	3	"In Ordnung."
Restarting the game action	1	"Soll ich wirklich neu starten? "
Restarting the game action	2	"Fehlgeschlagen."
Restoring the game action	1	"Laden des Spielstands fehlgeschlagen."
Restoring the game action	2	"In Ordnung."
requesting the pronoun meanings action	1	"Die Pronomen beziehen sich im Moment auf Folgendes:"
requesting the pronoun meanings action	2	": "
requesting the pronoun meanings action	3	": nicht gesetzt"
requesting the pronoun meanings action	4	"Dieses Spiel kennt keine Pronomen. (Schade!)"
requesting the pronoun meanings action	5	"."
requesting the score action	1	"[if the story has ended]In diesem Spiel [otherwise]Bislang [end if]habe ich [score] von [the maximum score] in [turn count] [if turn count is not 1]Zügen[otherwise]Zug[end if][ignore library line break]"
requesting the score action	2	"In diesem Spiel gibt es keine Punkte."
requesting the score action	3	", mit dem Rang "
Rubbing action	1	"Ich erreiche damit nichts."
Saving the game action	1	"Der Spielstand konnte nicht abgespeichert werden."
Saving the game action	2	"In Ordnung."
Saying No action	1	"Hey, das war eine rhetorische Frage."
Saying sorry action	1	"Schwamm drüber."
Saying Yes action	1	"Hey, das war eine rhetorische Frage."
Searching action	1	"Aber es ist dunkel."
Searching action	2	"Auf [dem noun] ist nichts."
Searching action	3	"Auf [dem noun] sehe ich [a list of things on the noun with accusative]."
Searching action	4	"Ich finde nichts Interessantes."
Searching action	5	"Ich kann nicht hineinschauen, [der noun] [ist] geschlossen."
Searching action	6	"[Der noun] [ist] leer."
Searching action	7	"In [dem noun] sehe ich [a list of things in the noun with accusative]."
Searching action	8	"[Der person asked] durchsuch[t] [den noun]."
Setting it To action	1	"Ich kann [den noun] nicht einstellen."
Showing it to action	1	"Aber ich habe [den noun] gar nicht."
Showing it to action	2	"[Der noun] [ist] nicht beeindruckt."
Singing action	1	"Ich singe. Nicht sehr schön."
Sleeping action	1	"Ich fühle mich nicht müde."
Smelling action	1	"Ich rieche nichts Ungewöhnliches."
Squeezing action	1	"La[ss] [deine] Hände bei [dir]."
Squeezing action	2	"Das bringt nichts."
Squeezing action	3	"[Der person asked] quetsch[t] [den noun]."
Swearing mildly action	1	"So, so."
Swearing obscenely action	1	"Hey, hör sofort damit auf."
Swinging action	1	"Dort gibt es nichts zu schaukeln."
Switching off action	1	"Ich kann [den noun] nicht ausschalten."
Switching off action	2	"[Der noun] ist bereits aus."
Switching off action	3	"Ich schalte [den noun] aus."
Switching off action	4	"[Der person asked] schalte[t] [den noun] aus."
Switching on action	1	"Ich kann [den noun] nicht einschalten."
Switching on action	2	"[Der noun] ist bereits an."
Switching on action	3	"Ich schalte [den noun] an."
Switching on action	4	"[Der person asked] schalte[t] [den noun] an."
switching score notification off action	1	"Meldungen bei Änderung des Punktestands aus."
switching score notification on action	1	"Meldungen bei Änderung des Punktestands ein."
switching the story transcript off action	1	"Es wird gar kein Protokoll mitgeschrieben."
switching the story transcript off action	2	"[line break]Ende des Protokolls."
switching the story transcript off action	3	"Der Versuch, das Protokoll zu schlie[ß]en, scheiterte."
switching the story transcript on action	1	"Es wird bereits ein Protokoll mitgeschrieben."
switching the story transcript on action	2	"Es wird nun ein Protokoll angelegt von"
switching the story transcript on action	3	"Der Versuch, ein Protokoll anzulegen, scheiterte."
Taking action	1	"In Ordnung."
Taking action	2	"Immer diese Selbstversessenheit."
Taking action	3	"Das würde [dem noun] bestimmt nicht gefallen."
Taking action	4	"Dazu mü[ss]te ich zunächst [if noun is a supporter]von[otherwise]aus[end if] [dem noun] [if noun is a supporter]herunter[otherwise]heraus[end if]."
Taking action	5	"Ich habe [den noun] bereits."
Taking action	6	"[Der noun] gehör[t] offenbar zu [dem library message object]."
Taking action	7	"[Der noun] [ist] offenbar ein Teil [the library message object]."
Taking action	8	"[Den library message object] gibt es hier nicht."
Taking action	9	"[Der library message object] [ist] nicht offen."
Taking action	10	"[Den library message object] kann ich nicht mitnehmen."
Taking action	11	"[Cap That's-They're library message object] fixed in place."
Taking action	12	"Du trägst bereits zu viele Dinge."
Taking action	13	"(Du verstaust [den library message object] in [dem current player's holdall] um Platz zu schaffen.)[ignore library line break]"
Taking action	14	"Du kannst nicht in [den library message object] hineingreifen."
Taking action	15	"Du kannst [den library message object] nicht tragen."
Taking action	16	"[Der person asked] [if the person asked is plural-named]nehmen[otherwise]nimmt[end if] [den noun]."
Taking inventory action	1	"Du hast nichts bei [dir]."
Taking inventory action	2	"Du hast [ignore library line break]"
Taking inventory action	3	"Folgendes bei [dir]:[line break]"
Taking inventory action	4	" bei [dir]."
Taking inventory action	5	"[Der person asked] durchsuch[t] [if the person asked is male]seine[otherwise]ihre[end if] Habe."
Taking off action	1	"Du trägst [den noun] gar nicht."
Taking off action	2	"Du ziehst [den noun] aus."
Taking off action	3	"[Der person asked] zieht [den noun] aus."
Tasting action	1	"Du schmeckst nichts Unerwartetes."
Telling it about action	1	"Du unterhältst [dich] ein wenig und erzählst [dir] alte Geschichten."
Telling it about action	2	"Keine Antwort."
Thinking action	1	"Gute Idee!"
throwing it at action	1	"Witzlos."
throwing it at action	2	"Im kritischen Augenblick fehlen @21 die Nerven dazu."
Tying it to action	1	"Dadurch würdest [du] nichts erreichen."
Touching action	1	"La[ss] [deine] Hände bei [dir]!"
Touching action	2	"Du fühlst nichts Unerwartetes."
Touching action	3	"Wenn [du] meinst."
Touching action	4	"[Der person asked] berühr[t] sich selbst."
Touching action	5	"[Der person asked] berühr[t] [dich]."
Touching action	6	"[Der person asked] berühr[t] [den noun]."
Turning action	1	"[Der noun] [ist] fest."
Turning action	2	"Du bist nicht dazu in der Lage."
Turning action	3	"Nichts passiert."
Turning action	4	"Das wäre sehr unhöflich."
Turning action	7	"[The person asked] turn[s] [the noun]."
Undressing action	1	"Du lä[ss]t [den noun] besser wie [er] [ist]."
Undressing action	2	"Besser nicht."
Undressing action	3	"[Der person asked] möchte[if person asked is plural-named]n[end if] gerne bleiben wie [er] [ist]."
Undressing action	4	"[Der person asked] möchte[if person asked is plural-named]n[end if] [den noun] nicht ausziehen."
Unlocking it with action	1	"Du kannst [den noun] nicht aufschlie[ß]en."
Unlocking it with action	2	"[Der noun] [ist] bereits aufgeschlossen."
Unlocking it with action	3	"[Der noun] pa[ss][t] nicht."
Unlocking it with action	4	"Du schlie[ß]t [den noun] auf."
Unlocking it with action	5	"[Der person asked] schlie[ß][t] [den noun] auf."
Verifying the story file action	1	"Die Spieldatei ist intakt."
Verifying the story file action	2	"Die Spieldatei ist korrupt."
Waiting action	1	"Die Zeit verstreicht."
Waiting action	2	"[Der person asked] warte[t]."
Waking action	1	"Das ist unnötig."
Waking up action	1	"Die bittere Wahrheit ist: Dies ist kein Traum."
Waving action	1	"Du hast [den noun] gar nicht."
Waving action	2	"Es sieht blöd aus, wie [du] mit [dem noun] wedelst."
Waving action	3	"[Der person asked] wink[t] mit [dem noun]."
Waving Hands action	1	"Du winkst und fühlst [dich] dabei etwas komisch."
Wearing action	1	"Du kannst [den noun] nicht anziehen!"
Wearing action	2	"Du hast [den noun] gar nicht!"
Wearing action	3	"Du trägst [den noun] bereits!"
Wearing action	4	"Du ziehst [den noun] an."
Wearing action	5	"[Der person asked] zieh[t] [den noun] an."



Section - General settings

Use no scoring.

The story genre is "Survival Horror".
The story headline is "Ein interaktiver Fluchtversuch".

When play begins:
	say "Ein leiser immer wiederkehrender Piepton ist zu hören; sonst nichts. Du schlägst langsam die Augen auf und schüttelst die Benommenheit so gut es geht ab. Was ist bloß passiert? Wie kommst du hierher?[paragraph break]Das Einzige, woran du dich momentan erinnern kannst, ist ein blendendes Licht, kurz vor diesem Aufprall. Du hattest keine Chance, überhaupt keine. Die Kurve war unübersichtlich und das andere Auto hatte gerade an dieser Stelle zum Überholen angesetzt. Das Letzte, was du bewusst wahrgenommen hast, war das Knirschen von Metall und das Splittern der Windschutzscheibe.[paragraph break]";
	say "Falls du mehr über diese Geschichte wissen möchtest, gib bitte INFO für eine kurze Information oder CREDITS für die Danksagung ein, Hinweise zur Lösung gibt es über HILFE. Noch nie Interactive Fiction gespielt? Gib RATGEBER ein.".

A thing is usually privately-named. A room is usually privately-named.

A thing can be moved.

A person can be asked.

Definition: a thing is other if it is not the player.

Instead of doing something other than examining to something which is part of a person:
	say "Dazu bist du nicht in der Lage.".

Instead of taking something which is carried by an other person (called the owner):
	say "[Er owner] ist sicher nicht gewillt, [den noun] herzugeben.".

Understand "kletter durch [something]" as climbing.

Understand "brich [something] auf" as attacking.

Section - Room-descriptors

Understand "Krankenhaus [n]" or "Klinik [f]" as the hospital.
Understand "Schacht [m]" or "Schachtende [n]" as the ventilation system.
Understand "Schacht [m]" or "Schachtende [n]" as the west end.
Understand "Hinterhof [m]" or "Hof [m]" as the backyard.
Understand "link" and "linker" and "links" and "Pfad [m]" and "Weg [m]" as the left path.
Understand "recht" and "rechter" and "rechts" and "Pfad [m]" and "Weg [m]" as the right path.
Understand "Lichtung [f]" as the clearing.
Understand "tot" and "Wald [m]" and "Waeldchen [n]" as the dying forest.
Understand "Eingangshof [m]" or "Hof [m]" or "Vorplatz [m]" or "Muellhalde [f]" as the dumping ground.
Understand "Strasse [f]" as the street.
Understand "Auto [n]" or "Wagen [m]" or "Automobil [n]" or "Polizeiauto [n]" or "Polizeiwagen [m]" or "Bullenschaukel [f]" as the police car.

Section - Conversation handling

Instead of telling someone about something, try asking the noun about it.

Instead of vaguely communicating:
	if the player can see the nurse:
		try answering the nurse that;
		stop the action;
	if the player can see the caretaker:
		try answering the caretaker that;
		stop the action;
	if the player can see the policeman:
		try answering the policeman that;
		stop the action;
	say "Du bist sprachlos.".

Vaguely asking is an action applying to one topic.

Understand "frag ueber/zu/nach [text]" as vaguely asking.
Understand "frag ueber/zu/nach [text]" as vaguely asking.
Understand "erzaehl ueber/von [text]" as vaguely asking.
Understand  "red ueber/von [text]" as vaguely asking.
Understand  "schwatz ueber/von [text]" as vaguely asking.
Understand  "schwätz ueber/von [text]" as vaguely asking.
Understand  "sprich ueber/von [text]" as vaguely asking.
Understand  "sprech ueber/von [text]" as vaguely asking.
Understand  "bericht ueber/von [text]" as vaguely asking.

Check vaguely asking:
	if the player can not see a person:
		say "Du redest ein wenig mit dir selbst. Das hat dich schon immer beruhigt.";
		stop the action.

Carry out vaguely asking:
	now the noun is a random other person in the location;
	try asking the noun about.

Section - Knowing relationship

Knowing about relates various people to various things. The verb to know about (he knows about, they know about, it is known about) implies the knowing about relation.

Section - Faraway

A thing can be faraway.

Before doing something to a thing which is faraway:
	if the current action is trying rubbing something or squeezing something or burning something or cutting something or tying something to something or swinging something or taking something or touching something or smelling something or pulling something or pushing something or turning something or opening something or closing something or unlocking something with something:
		if the noun is the ceiling or the noun is the hole:
			say "Von hier kannst du [den noun] nicht erreichen, es sei denn, du steigst auf das Bett.";
			stop the action;
		if the noun is the table:
			say "Der Tisch ist zu weit unten, das einzige, was sich in deiner Reichweite befindet, ist der Schrank.";
			stop the action;
		if the noun is the station-cabinet:
			say "Du könntest den Schrank erreichen, wenn das Gitter geöffnet wäre.";
			stop the action;
		say "[Der noun] [ist] zu weit entfernt.";
		stop the action.

Part - The main act

Section - World inside

The indoors is a region.

The hospital is a neuter room, in the indoors. "[if unvisited]Wo bist du? Soll das hier etwa ein Krankenhaus sein? Zunächst schaut der Raum eher aus wie eine düstere, unaufgeräumte Abstellkammer.[otherwise]Dieser Raum ist eng und bedrückend, verwahrlost und schmutzig. Gerümpel ist quer über den ganzen Raum verstreut. Die Möblierung besteht aus einem Bett, besser gesagt einer unbequemen, harten Liege und einem rostigen Metallschrank.[paragraph break]Eine Tür befindet sich im Süden und ein kleines vergittertes Fenster im Norden.".

The printed name of the hospital is "Krankenhaus".

The description of the player is "Dein Körper ist über und über mit kleinen Schnittwunden übersät, die wahrscheinlich von dem Unfall stammen. Du fühlst dich wie gerädert.".

Understand "schnittwunden" or "wunden" as yourself.

The rubbish is neuter scenery in the hospital. The printed name of the rubbish is "nutzlos[^] Gerümpel[-s]". Understand "nutzlos" and "geruempel" and "sachen [p]" as the rubbish. The description of the rubbish is "Lauter komplett nutzlose Sachen, die kreuz und quer im Raum verteilt sind.".

The special indefinite article of the rubbish is NO ARTICLE.

Instead of doing anything other than examining to the rubbish:
	say "Es macht wenig Sinn, sich mit nutzlosem Gerümpel zu beschäftigen.".

The wall is female plural-named scenery in the hospital. The printed name of the wall is "massiv[^] Wände". Understand "massiv" and "Wand" and "waende [p]" as the wall. The description of the wall is "Massive Wände umgeben diesen kleinen Raum.".

Instead of attacking the wall:
	say "Du hämmerst wie wild auf die Wand ein - nutzlos.".

The ceiling is female scenery in the hospital. The printed name of the ceiling is "Kassettendecke". Understand "weiss" and "kassetten [p]" and "Decke" and "paneele [p]" and "deckenpaneele [p]" and "kassettendecke" as the ceiling. The description of the ceiling is "Eine Kassettendecke aus weißen Paneelen[if the hole is in the hospital] mit einem Loch, das in einen Lüftungsschacht führt[end if].".

Definition:
	The ceiling is faraway if the player is not on the bed.

Instead of touching or turning or pushing or pulling or attacking the ceiling when the player is on the bed:
	if the hole is in the hospital:
		say "Das wird wenig bringen.";
	otherwise:
		say "Du betastest hastig alle Deckenpaneele - plötzlich gibt eins nach. Du schiebst es ganz zurück: ein Loch ist entstanden, durch das du hindurchpassen könntest. Es führt in einen kleinen Lüftungsschacht.";
		move the hole to the hospital.

Instead of opening the ceiling:
	try touching the ceiling.

The hole is a neuter thing. "In der Decke ist ein Loch zu sehen, das in einen kleinen Lüftungsschacht führt.". The printed name of the hole is "dunkl[^] Schacht[-s]". Understand "dunkel" and "Loch" and "Schacht [m]" as the hole. The description of the hole is "Der Schacht ist ziemlich dunkel.".

Definition:
	The hole is faraway if the player is not on the bed.

Instead of entering the hole when the player is not on the bed:
	say "Von hier kannst du den Schacht nicht erreichen, aber vielleicht wenn du auf das Bett steigst.".

Instead of entering the hole when the player is on the bed:
	say "Du hangelst dich hoch und ziehst dich mit letzter Kraft in den Schacht hinein.";
	show-left figure shaft;
	move the player to the ventilation system.

Instead of going up from hospital when the hole is not in the hospital:
	say "Du hast keine Ahnung, wie.".
	
Instead of going up from hospital when the hole is in the hospital:
	try entering the hole.

The bed is a neuter scenery enterable supporter in the hospital. The printed name of the bed is "alt[^] Bett[-s]". Understand "alt" and "Bett" and "Liege [f]" and "Sperrholz [n]" as the bed. The description of the bed is "Das Bett ist ein dünne Liege aus Sperrholz. [if crossways]Es nimmt fast den gesamten Raum zwischen Schrank und gegenüberliegender Wand ein, so dass die Schranktür unmöglich zu öffnen ist[otherwise]Es nimmt fast den gesamten Raum zwischen Tür und Fenster ein[end if].".

The bed is either crossways or lengthways. The bed is crossways.

The player is on the bed.

The cabinet is a male enterable scenery closed and openable container in the hospital. The printed name of the cabinet is "rostig[^] Metallschrank[-s]". Understand "rostig" and "Schrank" and "Metallschrank" and "Beulen [p]" and "Rost [m]" and "Roststellen [p]" as the cabinet. The description of the cabinet is "Der Schrank sieht ziemlich heruntergekommen aus. Überall sind Roststellen und Beulen in der Oberfläche. [if closed]Der Schrank ist im Moment geschlossen[otherwise]Der Schrank ist geöffnet[end if].".

The white door is a female scenery door, south of the hospital and north of the corridor. The printed name of the white door is "weiß[^] Tür".  Understand "weiss" and "Tuer" as the white door. The white door is closed. The description of the white door is "Eine weiße, stabil aussehende Tür, die - unüblich für ein Krankenhaus und seine vorgeschriebenen Rettungswege - sich nach innen öffnet.".

Instead of unlocking the door with the crowbar:
	say "Du bist eigentlich froh, dass sie im Momemt verschlossen ist.";

The window is a neuter scenery door, north of the hospital. The printed name of the window is "klein[^] Fenster[-s]".  Understand "klein" and "vergittert" and "Fenster" and "Gitter [n]" as the window. The window is closed. The description of the window is "Ein kleines, vergittertes Fenster in der Außenwand.".

Instead of pushing or pulling the bed for the first time:
	say "Es ist so eng hier, dass das Bett nur wenige Zentimeter verschoben werden kann - es passt nahezu exakt in diesen fast quadratischen Raum hinein.".

Instead of pushing or pulling the bed:
	say "Es ist zwar sehr eng hier, aber vielleicht könnte man das Bett drehen, es scheint nicht allzu schwer zu sein.".

Instead of turning the bed:
	if the player is on the bed:
		say "Das geht nicht solange du auf dem Bett liegst.";
		stop the action;
	if the white door is open:
		say "Die offene Tür verhindert jegliches Bewegen des Betts.";
		stop the action;
	if the cabinet is open:
		say "Der offene Metallschrank verhindert jegliches Bewegen des Betts.";
		stop the action;	
	say "Mit großer Mühe gelingt es dir, das Bett in eine Schräglage zu bringen, zu drehen und es wieder abzustellen. [if the bed is crossways]Es blockiert jetzt die Tür[otherwise]Es blockiert jetzt den Schrank[end if].";
	now the bed is moved;
	if the bed is lengthways:
		now the bed is crossways;
	otherwise:
		now the bed is lengthways.

Instead of going down when the player is on something:
	try exiting.

Instead of opening the white door when the bed is lengthways:
	say "Das alte Bett blockiert die Tür. Sie kann momentan nicht geöffnet werden.".

Instead of opening the cabinet when the bed is crossways:
	say "Das alte Bett blockiert den Schrank. Er kann momentan nicht geöffnet werden.".
	
Instead of opening the white door when the player is on the bed:
	say "Dazu musst du erst aufstehen.".

Instead of closing the white door when the nurse is in the hospital:
	say "[The nurse] blockiert die Tür. Du kannst sie nicht schliessen.".

Instead of going south when location is hospital:
	if the player is on the bed:
		say "Dazu musst du erst aufstehen.";
		stop the action;
	if the white door is closed:
		say "Dazu musst du erst die Tür öffnen.";
		stop the action;
	if the nurse is not in the hospital:
		move the nurse to the hospital;
		if advisor is switched on:
			add the nurse to advise list;
			repeat with item running through things worn by the nurse:
				add item to advise list;
			repeat with item running through things carried by the nurse:
				add item to advise list;			
			repeat with item running through things which are part of the nurse:
				add item to advise list;
		now the player knows about the nurse;
		stop the action;
	otherwise:
		say "Du blickst sehnsüchtig auf die Tür. Eins ist todsicher: sie wird dich nicht gehen lassen.".

Instead of entering the door:
	if the player is on the bed:
		say "Dazu musst du erst aufstehen.";
		stop the action;
	if the white door is closed:
		say "Dazu musst du erst die Tür öffnen.";
		stop the action;
	if the nurse is not in the hospital:
		move the nurse to the hospital;
		if advisor is switched on:
			add the nurse to advise list;
			repeat with item running through things worn by the nurse:
				add item to advise list;
			repeat with item running through things carried by the nurse:
				add item to advise list;			
			repeat with item running through things which are part of the nurse:
				add item to advise list;
		now the player knows about the nurse;
		stop the action;
	otherwise:
		say "Du blickst sehnsüchtig auf die Tür. Eins ist todsicher: sie wird dich nicht gehen lassen.".

Instead of going north when location is hospital:
	if the player is on the bed:
		say "Dazu musst du erst aufstehen.";
		stop the action;
	if the window is closed:
		say "Das Fenster ist vergittert. Keine Chance.";
		stop the action;
	if advisor is switched on:
		now advisor is switched off;
		say "(Ratgeber Modus ist deaktiviert)[italic type][paragraph break]So weit, so gut. Du bist also aus dem Krankenhaus entkommen. Ich muss leider hier bleiben. Viel Glück, ich drücke dir die Daumen![roman type].";
	say "Du kletterst durch das Fenster hinaus.";
	move the player to the backyard;
	show-left figure backyard.

Instead of sleeping when the player is not on the bed:
	try entering the bed.

Instead of sleeping:
	say "Das Letzte, woran du jetzt denken kannst, ist Schlaf.".

Instead of jumping when the player is on the bed:
	say "Du springst ein wenig auf der Matratze.".

The crowbar is a female thing in the cabinet. The printed name of the crowbar is "schwer[^] Brechstange". Understand "schwer" and "Stange" and "Brechstange" and "Brecheisen" as the crowbar. The description of the crowbar is "Eine schwere Brechstange aus massivem Stahl.".

Section - Ventilation system

Ventilation system is a male room, above the hospital, in the indoors. "Hier ist es sehr eng und stickig und es ist so gut wie nichts erkennbar. Etwas weiter westlich endet der Schacht vor einem Gitter, durch das von unten Licht herein fällt.[paragraph break]Der einzige Ausgang ist das Loch, das zurück in dein Zimmer führt."

The printed name of the ventilation system is "Lüftungsschacht".

Instead of going down from ventilation system:
	say "Vorsichtig hangelst du dich wieder nach unten auf das Bett.";
	show-left figure title;
	move the player to the bed.

The ventilation-hole is a neuter scenery in the ventilation system. The printed name of the ventilation-hole is "Loch[-s]". Understand "Loch" as the ventilation-hole. The description of the ventilation-hole is "Das Loch ist gerade groß genug, um hindurch zu passen.".

Instead of going nowhere from the ventilation system:
	say "Die Schachtwand ist im Weg."

Instead of exiting when location is ventilation system:
	try going down.

Section - Westend

West end is a neuter room, west from the ventilation system, in the indoors. "Durch einen kleines Gitter im Boden fällt etwas Licht in den Schacht. Unter dir befindet sich das Stationszimmer. Der Raum ist mit einem Tisch und einem Schrank möbliert und sieht genauso dreckig aus wie dein Zimmer.[paragraph break]Die einzige Richtung, in die du dich von hier bewegen kannst, ist zurück nach Osten."

The printed name of the west end is "Schachtende".

The grate is a neuter scenery openable and closed container in the west end. The printed name of the grate is "Lüftungsgitter". Understand "gitter" or "lueftungsgitter" as the grate. the description of the grate is "Ein kleines Lüftungsgitter aus Metall, das im Moment [if open]geöffnet[otherwise]geschlossen[end if] ist."

Instead of dropping something when location is west end:
	say "Du möchtest [den noun] jetzt keinesfalls hier ablegen.".

before taking something when location is west end:
	if the grate is closed:
		say "Dazu musst du erst das Lüftungsgitter öffnen.";
		stop the action.

The station-cabinet is a male enterable scenery closed and openable container in the west end. The printed name of the station-cabinet is "rostig[^] Metallschrank[-s]". Understand "rostig" and "Schrank" and "Metallschrank" and "Beulen [p]" and "Rost [m]" and "Roststellen [p]" as the station-cabinet. The description of the station-cabinet is "Der Schrank schaut genauso aus, wie der in deinem Zimmer. Überall sind Roststellen und Beulen in der Oberfläche. [if closed]Der Schrank ist im Moment geschlossen[otherwise]Der Schrank ist geöffnet[end if].".

Definition:
	The station-cabinet is faraway if the grate is closed.

Report opening the station-cabinet:
	say "Durch das offene Gitter erreichst du mit Mühe die Kante der Schranktür.".

The table is a male scenery in the west end. The printed name of the table is "weiss[^] Tisch[-s]". Understand "weiss" and "Tisch" and "flecken [p]" as the table. The description of the table is "Der Tisch ist weiß und glatt und seine Oberfläche weist seltsame Flecken auf.".

The table is faraway.

The dummy room is below the west end.

Instead of going down from west end:
	say "Selbst wenn das Lüftungsgitter geöffnet ist - du passt niemals durch das Loch hindurch."

Instead of going nowhere from the west end:
	say "Die Schachtwand ist im Weg."

Instead of exiting when location is west end:
	try going down.

Section - The window

Instead of attacking or pushing or pulling the window:
	if the player is not holder of the crowbar:
		if the player can see the crowbar:
			say "Mit dieser Brechstange ließe sich das Fenster sicher öffnen.";
		otherwise:
			say "Mit den bloßen Händen wirst du nicht weiterkommen.";
	otherwise:
		try unlocking the window with the crowbar.

Instead of unlocking the window with the crowbar when location is hospital:
	if the window is open:
		say "Das hast du doch bereits getan.";
		stop the action;
	if the player does not know about the nurse:
		say "Bevor du fremdes Eigentum beschädigst, willst du erst wissen, was hinter dieser weißen Tür ist.";
		stop the action;
	say "Mit etwas Anstrengung verbiegst du das Fenstergitter soweit, dass du durch die entstandene Öffnung hinausklettern kannst.";
	now the window is open.

Instead of opening the window:
	say "Mit den bloßen Händen kommst du hier nicht weiter.".

Instead of unlocking the window with the crowbar when location is backyard:
	say "Aber das Fenster ist schon geöffnet.";

Instead of entering the window when location is backyard:
	try going south.

Instead of climbing the window when location is backyard:
	try going south.

Instead of entering the window when location is hospital:
	try going north.

Instead of climbing the window when location is hospital:
	try going north.

Does the player mean doing something to the window:
	it is likely.

Section - World outside

The outdoors is a region.

The fence is male scenery backdrop in the outdoors. The printed name of the fence is "Zaun[-s]". Understand "zaun" or "einzaeunung [f]" or "hindernis [n]" as the fence. The description of the fence is "Ein hoher, unüberwindbarer Zaun.".

Instead of entering or climbing the fence:
	say "Der Zaun ist viel zu hoch.".

Section - Backyard

The backyard is a male room, north from the window, in the outdoors. "[if unvisited]Noch bist du nicht in Sicherheit, das ganze Gelände ist von einem hohen Zaun umschlossen. Es scheint, als liegt der Vordereingang auf der anderen Seite.[paragraph break][end if]Du stehst vor einem riesigen schwarzen Gebäude mit hunderten dieser kleinen, vergitterten Fenster. Der Hinterhof ist schmutzig, verwahrlost und in allen Ecken liegt Müll.[paragraph break]Zwei Wege umrunden das Gebäude, einer führt nach Osten und einer nach Westen.".

The printed name of the backyard is "Hinterhof".

Instead of going south when location is backyard:
	say "Niemals.".

The backyard-building is a neuter scenery in the backyard. The printed name of the backyard-building is "schwarz[^] Gebäude[-s]". Understand "schwarz" and "Gebaeude" and "Bauwerk" and "Haus" and "Kirche [f]" and "Fenster [p]" and "Gitter [p]" as the backyard-building. the description of the backyard-building is "Das ganze Gebäude ist schwarz gestrichen und hebt sich kaum vom Nachthimmel ab. Hunderte dieser vergitterten Fenster sind scheinbar wahllos über die gesamte Fassade verteilt.".

The backyard-rubbish is male scenery in the backyard. The printed name of the backyard-rubbish is "Müll". Understand "Muell" or "Unrat" or "Dreck" or "Abfall" as the backyard-rubbish. The description of the backyard-rubbish is "Lauter Müll und Unrat, verstreut bis in die hintersten Ecken.".

Does the player mean examining the backyard-building:
	it is very likely.

Section - Left path

The left path is a male room, west from the backyard, in the outdoors. "Das Außengelände scheint ziemlich groß zu sein. Die Einzäunung ist noch schwach in der Ferne erkennbar. Der Pfad ist links und rechts von dichter Bepflanzung gesäumt.[paragraph break]Im Süden liegt ein kleines Wäldchen und im Osten geht es zurück zum Hinterhof.".

The printed name of the left path is "link[^] Pfad".

The greenery is female scenery in the left path. The printed name of the greenery is "Bepflanzung". Understand "Bepflanzung" or "Gruen [n]" or "Pflanzen [p]" or "Buesche [p]" or "Dornen [p]" as the greenery. The description of the greenery is "Dichte Bepflanzung mit Dornen, auf beiden Seiten des Wegs.".

Section - Dying forest

The dying forest is a male room, south from the left path, in the outdoors. "Unheimlich und bizarr zeichnen sich die kargen Umrisse der kahlen und knorrigen Äste gegen den Nachthimmel ab. Obwohl es Hochsommer ist, wächst hier kein Grün. Die Bäume sind abgestorben und der Boden ist über und über mit welken Blättern bedeckt.[paragraph break]Im Norden liegt der Pfad zum Hinterhof und südlich ist eine kleine Lichtung erkennbar.".

The printed name of the dying forest is "tot[^] Wald".

The ground-leaves are a female plural-named scenery in the dying forest. The printed name of the ground-leaves is "Blätter". Understand "Blaetter" or "Boden [m]" or "Laub [n]" as the ground-leaves. The description of the ground-leaves is "Die grauen Blätter bedecken den gesamten Boden hier.".

Instead of taking the ground-leaves when the leaves are in the pit:
	if the leaves are in the pit:
		say "Du nimmst eine Handvoll Blätter mit.";
		move the leaves to the player;
	otherwise:
		say "Du hast bereits eine Handvoll davon mitgenommen.".

Instead of dropping the leaves when location is dying forest:
	say "Du wirfst die Blätter wieder auf den Boden.";
	move the leaves to the pit.

The branches is plural-named scenery in the dying forest. The printed name of the branches is "Äste". Understand "tote" and "Aeste" and "Baeume" as the branches. The description of the branches is "Die Bäume sind grau und verdorrt und wirken wie ein Relikt aus besseren Tagen.".

Section - Clearing

The clearing is a female room, south from the dying forest, in the outdoors. "Diese Lichtung ist ringsum mit toten Bäumen umgeben. In der Mitte der Lichtung befindet sich ein runder Holzdeckel, etwa zwei Meter im Durchmesser, umgeben von einem Rand aus gemauertem Stein.[paragraph break]Der einzige Weg führt von hier zurück nach Norden.".

The printed name of the clearing is "Lichtung".

The trees is plural-named scenery in the clearing. The printed name of the trees is "Äste". Understand "tote" and "Aeste" and "Baeume" as the trees. The description of the trees is "Die Bäume sind grau und verdorrt und wirken wie ein Relikt aus besseren Tagen.".

The edge is male scenery in the clearing. The printed name of the edge is "gemauert[^] Rand[-s]". Understand "gemauert" and "Rand" and "Steine [p]" or "Einfassung [f]" as the edge. The description of the edge is "Eine gemauerte Einfassung, um den Holzdeckel.".

A trapdoor is a kind of supporter. A trapdoor is either open or closed. A trapdoor is always enterable.

The wooden trapdoor is a male scenery trapdoor in the clearing. It is closed. The printed name of the wooden trapdoor is "Holzdeckel[-s]". understand "Falltuer [f]" or "Holzdeckel" or "Deckel" or "Fuge [f]" as the wooden trapdoor.

The lever is male scenery in the clearing. The printed name of the lever is "Hebel[-s]". Understand "hebel" as the lever. The description of the lever is "Ein Hebel aus rostigen Metall, seitlich neben dem hölzernen Deckel angebracht.". The description of the wooden trapdoor is "[if closed]Der Holzdeckel verdeckt irgendetwas. In der Mitte teilt eine schmale Fuge den Deckel in zwei Hälften.[otherwise]Der Holzdeckel ist geöffnet und gibt einen schwarzen Schacht, der in die Tiefe führt, frei.[end if] Seitlich befindet sich ein Hebel daran.".

The lever is either pushed or pulled. The lever is pushed.

The shaft is a male scenery. the shaft is in the pit. The printed name of the shaft is "schwarz[^] Schacht[-s]". Understand "schwarz" and "schacht" as the shaft. The description of the shaft is "Der Schacht ist schwarz und du kannst keinen Boden sehen. Irgendetwas scheint sich dort unter aber zu bewegen.".

Instead of searching the shaft:
	say "De Schacht ist schwarz und anscheinend sehr tief. Ganz weit unten kannst du schwach Bewegungen erkennen - oder hast du dich getäuscht? Du siehst jedenfalls keine Leiter, und wenn du hinunter gelangen willst, bleibt dir keine andere Wahl als zu springen.".

Instead of jumping when the location is the clearing and the wooden trapdoor is open:
	try going down.

For printing a locale paragraph about the wooden trapdoor:
	set the locale priority of the wooden trapdoor to 0; 
	if the leaves are on the wooden trapdoor and the crowbar is on the wooden trapdoor:
		say "Der hölzerne Deckel in der Mitte der Lichtung und die Brechstange darauf sind unter einer Schicht Laub perfekt verborgen.";
		now the leaves are mentioned;
		now the crowbar is mentioned;
		stop the action;
	if the leaves are on the wooden trapdoor:
		say "Der hölzerne Deckel in der Mitte der Lichtung ist unter einer Schicht Laub perfekt verborgen.";
		now the leaves are mentioned;
		stop the action;
	if the crowbar is on the wooden trapdoor:
		say "Auf dem hölzernen Deckel in der Mitte der Lichtung liegt eine Brechstange.";
		now the crowbar is mentioned;
		stop the action;
	if the shaft is in the clearing::
		say "Ein schwarzer Schacht führt in die Tiefe.";
		stop the action.

Instead of opening the wooden trapdoor:
	if the wooden trapdoor is closed:
		say "Du rüttelst ein wenig an dem hölzernen Deckel. Er bewegt sich keinen Millimeter.";
	otherwise:
		say "Der Deckel ist doch bereits geöffnet.".

Instead of closing the wooden trapdoor:
	if the wooden trapdoor is open:
		say "Du ziehst mit aller Kraft an dem hölzernen Deckel. Er bewegt sich keinen Millimeter.";
	otherwise:
		say "Der Deckel ist doch bereits geschlossen.".

Before putting the leaves on the wooden trapdoor:
	if the wooden trapdoor is open:
		say "Die Blätter fallen in den offenen Schacht.";
		move the leaves to the pit;
		stop the action.

Before putting the crowbar on the wooden trapdoor:
	if the wooden trapdoor is open:
		say "Die Brechstange fällt in den offenen Schacht. Du hörst einen klirrenden Aufprall.";
		move the crowbar to the pit;
		stop the action.

Instead of inserting the leaves into the shaft:
	say "Die Blätter fallen in den offenen Schacht.";
	move the leaves to the pit.

Instead of inserting the crowbar into the shaft:
	say "Die Brechstange fällt in den offenen Schacht. Du hörst einen klirrenden Aufprall.";
	move the crowbar to the pit.

Instead of entering the wooden trapdoor when the trapdoor is open:
	say "Der Holzdeckel ist geöffnet und kann nicht betreten werden.".

Report putting the leaves on the wooden trapdoor:
	say "Du verbirgst den Holzdeckel vollständig unter einer Blätterschicht.";
	stop the action.

Instead of pulling or pushing the lever:
	if the lever is pushed:
		now the lever is pulled;
		now the wooden trapdoor is open;
		say "Mit einem quietschenden Geräusch bewegst du den Hebel in eine Richtung; der hölzerne Deckel springt wie eine Falltür nach unten auf und gibt einen tiefen Schacht frei.";
		move the shaft to the clearing;
		if the player is on the wooden trapdoor and the caretaker is on the wooden trapdoor:
			say "[line break]Ihr stürzt beide in die Tiefe. Du fällst ein gutes Stück nach unten, absolute Dunkelheit um dich herum. Du landest verhältnismäßig weich, aber deine anfängliche Erleichterung wandelt sich in pures Entsetzen, als du bemerkst, dass zahlreiche Arme nach dir greifen und dich weiter in die Tiefe ziehen.";
			end the story finally saying "Du bist verloren.";
		if the player is on the wooden trapdoor:
			say "[line break]Du stürzst in die Tiefe. Du fällst ein gutes Stück nach unten, absolute Dunkelheit um dich herum. Du landest verhältnismäßig weich, aber deine anfängliche Erleichterung wandelt sich in pures Entsetzen, als du bemerkst, dass zahlreiche Arme nach dir greifen und dich weiter in die Tiefe ziehen.";
			end the story finally saying "Du bist verloren.";
		if the caretaker is on the wooden trapdoor:
			say "[line break]Der Koloss schreit auf und stürzt in die Tiefe - kurz darauf folgt ein dumpfer Aufprall. Du schaust nach unten, kannst aber in der Dunkelheit nichts erkennen.";
			move the caretaker to the pit;
		if the leaves are on the wooden trapdoor and the crowbar are on the wooden trapdoor:
			say "[line break]Die Blätter auf dem Deckel fallen zusammen mit der Brechstange in die Tiefe.";
			move the leaves to the pit;	
			move the crowbar to the pit;
			stop the action;
		if the leaves are on the wooden trapdoor:
			say "[line break]Die Blätter auf dem Deckel fallen langsam in die Tiefe.";
			move the leaves to the pit;
		if the crowbar is on the wooden trapdoor:			
			say "Die Brechstange fällt in den offenen Schacht. Du hörst einen klirrenden Aufprall.";
			move the crowbar to the pit;
		stop the action;
	if the lever is pulled:
		now the lever is pushed;
		now the wooden trapdoor is closed;
		move the shaft to the pit;
		say "Mit einem quietschenden Geräusch bewegst du den Hebel wieder zurück; der hölzerne Deckel schliesst sich wieder.".

Instead of going down when location is clearing:
	if the trapdoor is open:
		say "Verzweifelt springst du in die Tiefe. Du fällst ein gutes Stück nach unten, absolute Dunkelheit um dich herum.. Du landest verhältnismäßig weich, aber deine anfängliche Erleichterung wandelt sich in pures Entsetzen, als du bemerkst, dass zahlreiche Arme nach dir greifen und dich weiter in die Tiefe ziehen.";
		end the story finally saying "Du bist verloren.";
	otherwise:
		say "Du hast keine Ahnung, wie du das bewerkstelligen sollst.".

Instead of entering the shaft:
	try going down.

Section - Pit

The pit is a room, below the clearing.

The leaves are a female plural-named thing in the pit. The printed name of the leaves is "Blätter". The indefinite article of the leaves is "einige". Understand "einige" and "blaetter" as the leaves. The description of the leaves is "Eine Handvoll Blätter. Trocken und grau.".

Section - Right path

The right path is a  male room, east from the backyard, in the outdoors. "Der Pfad ist genauso verdreckt wie der Hinterhof und du musst dich anstrengen, um auf keinen Abfall zu treten. Das schwarze Gebäude besitzt auch auf dieser Seite zahlreiche vergitterte Fenster.[paragraph break]Im Süden liegt der Eingangshof und das Tor in der Einzäunung ist von hier aus gut zu sehen. Im Westen geht es zurück zum Hinterhof.".

The printed name of the right path is "recht[^] Pfad".

The path-building is a neuter scenery in the right path. The printed name of the path-building is "schwarz[^] Gebäude[-s]". Understand "schwarz" and "Gebaeude" and "Bauwerk" and "Haus" and "Kirche [f]" and "Fenster [p]" and "Gitter [p]" as the path-building. the description of the path-building is "Das ganze Gebäude ist schwarz gestrichen und hebt sich kaum vom Nachthimmel ab. Hunderte dieser vergitterten Fenster sind scheinbar wahllos über die gesamte Fassade verteilt.".

The path-rubbish is male scenery in the right path. The printed name of the path-rubbish is "Müll[-s]". Understand "Muell" or "Unrat" or "Dreck" or "Abfall" as the path-rubbish. The description of the path-rubbish is "Lauter Müll und Unrat, verstreut bis in die hintersten Ecken.".

Instead of doing anything other than examining to the dumping-rubbish:
	say "Es macht wenig Sinn, sich mit dem Müll zu beschäftigen.".

The far-gate is neuter scenery in the right path. The printed name of the far-gate is "Tor[-es]". Understand "Tor" or "Eingangstor" or "Eingang [m]" as the far-gate. The description of the far-gate is "Deine Fahrkarte in die Freiheit, ein großes offenes Tor in der Einzäunung, ein Stück weiter südlich.".

The far-gate is faraway.

Section - Dumping ground

The dumping ground is a male room, south from the right path, in the outdoors. "Der Eingangshof ist genauso verdreckt wie der Hinterhof. Überall liegt Unrat im Weg und eine große Mülltonne steht am Rand des Geländes. Auf dieser Seite sind zwei mächtige Türme aus der Fassade des schwarzen Gebäudes modelliert - fast wie bei einer Kirche oder besser Kathedrale.[paragraph break]Der Weg in die Freiheit durch das offene Eingangstor im Süden liegt vor dir[if the caretaker is in the dumping ground], wenn nicht ein Koloss von einem Mann direkt davor stehen würde[end if].[paragraph break]Im Norden führt der verdreckte Pfad zum Hinterhof zurück.".

The printed name of the dumping ground is "Eingangshof".

The dumping-building is a neuter scenery in the dumping ground. The printed name of the dumping-building is "schwarz[^] Gebäude[-s]". Understand "schwarz" and "Gebaeude" and "Bauwerk" and "Haus" and "Kirche [f]" and "Fenster [p]" and "Gitter [p]" and "Kathedrale [f]" as the dumping-building. the description of the dumping-building is "Das ganze Gebäude wirkt tatsächlich wie eine schwarze Kathedrale - die Türme ragen bedrohlich in den Nachthimmel.".

The dustbin is a female scenery closed and openable container in the dumping ground. The printed name of the dustbin is "riesig[^] Mülltonne". Understand "riesig" and "Muell" and "Tonne" and "Muelltonne" as the dustbin. the description of the dustbin is "Eine immens große Mülltonne steht hier am Zaun.".

Instead of opening the dustbin:
	say "Je länger du darüber nachdenkst, desto weniger willst du wissen, was wirklich in dieser Mülltonne steckt.".

The gate is a neuter scenery in the dumping ground. The printed name of the gate is "Eingangstor[-es]". Understand "Eingang" or "Tor" or "Eingangstor" as the gate. the description of the gate is "Das Tor ist glücklicherweise offen.".

Instead of opening or closing the gate:
	say "Du solltest dich jetzt nicht lange mit dem Tor beschäftigen.".

Instead of entering the gate:
	try going south.

The dumping-rubbish is male scenery in the dumping ground. The printed name of the dumping-rubbish is "Müll[-s]". Understand "muell" or "unrat" or "dreck" or "abfall" as the dumping-rubbish. The description of the dumping-rubbish is "Lauter Müll und Unrat, verstreut bis in die hintersten Ecken.".

Instead of doing anything other than examining to the dumping-rubbish:
	say "Es macht wenig Sinn, sich mit dem Müll zu beschäftigen.".

The towers are a plural-named scenery in the dumping ground. The printed name of the towers is "Türme". Understand "turm [m]" or "tuerme" as the towers. the description of the towers is "Die beiden Türme besitzen eine stattliche Höhe, zwischen ihnen ist die für eine Kathedrale typische Fensterrosette angebracht.".

The towers are faraway.

The rosette is a female scenery in the dumping ground. The printed name of the rosette is "Fensterrosette". Understand "Rosette" or "Fensterrosette" or "Fenster [n]" or "Glasfenster [n]" as the rosette. the description of the rosette is "Die Rosette ist ein farbig bemaltes Glasfenster, das eine schauderliche Szene zeigt. Untote erheben sich aus ihren Gräbern und stellen den Menschen nach. Sie werfen die Lebenden in einen tiefen Schacht und gießen brennendes Öl hinein.".

The rosette is faraway.

Section - Street

After going to the street:
	say "Du sammelst deine letzten Kräfte und rennst durch das Tor in die schützende Nacht hinaus. Nach einer guten Viertelstunde erreichst du schließlich die Strasse und atmest erleichtert auf. Ein Polizeiwagen kommt näher, wird langsamer und hält an. Es scheint, als hat das Glück dich nicht verlassen.";
	show-left figure street;
	try looking.

The printed name of the street is "klein[^] Strasse".

Instead of going north when location is street:
	say "Nicht um alles in der Welt willst du dorthin zurück."

Street is a female room, south from dumping ground, in the outdoors. "Eine kleine Nebenstrasse führt hier vorbei. Sie ist schwach durch eine Laterne beleuchtet, in deren Licht ein Polizeiwagen mit laufendem Motor wartet.[paragraph break]Ein gutes Stück weiter nördlich ragt dein ehemaliges Gefängnis schwarz und bedrohlich in den Nachthimmel.".

The street-car is a male scenery in the street. The printed name of the street-car is "Polizeiwagen[-s]". Understand "Wagen" or "Polizeiwagen" or "Auto [n]" or "Polizeiauto [n]" or "Automobil [n]" or "Bullenschaukel [f]" or "Wanne [f]" as the street-car. the description of the street-car is "Ein grau-grüner Polizeiwagen, mit schwarzen undurchsichtigen Scheiben. Du musst wohl einsteigen, um zu sehen, wer darin sitzt.".

The windscreen is a female scenery in the street. The printed name of the windscreen is "Windschutzscheibe". Understand "Scheibe" or "Windschutzscheibe" or "Scheiben [p]" as the windscreen. the description of the windscreen is "Schwarz und undurchsichtig, wie die Nacht.".

The lantern is female scenery in the street. The printed name of the lantern is "Laterne". Understand "Laterne" or "Licht [n]" or "Lampe" as the lantern. The description of the lantern is "Eine kleine Strassenlaterne.".

The street-building is a neuter scenery in the street. The printed name of the street-building is "schwarz[^] Gebäude[-s]". Understand "schwarz" and "Gebaeude" and "Bauwerk" and "Haus" and "Kirche [f]" and "Fenster [p]" and "Gitter [p]" and "Kathedrale [f]" and "Gefaengnis" as the street-building. the description of the street-building is "Du wirfst einen letzten Blick auf dein Gefängnis: die Türme ragen bedrohlich in den Nachthimmel.".

The street-building is faraway.

Instead of entering the street-car:
	try going inside.

Report dropping the crowbar when location is street:
	say "Möglichst unauffällig lässt du die Brechstange ins Gras fallen.";
	stop the action.

Understand "steig ein in [something]" as entering.

Does the player mean entering the street-car: it is likely.

Section - Police car

The police car is a neuter room, inside from the street. "Dieser Wagen ist einer von diesen typischen Polizeiautos, mit grauer und vergilbter Innenausstattung und einem starken Geruch nach Rauch.".

The printed name of the police car is "Im Polizeiauto".

Instead of exiting when location is police car:
	try going outside.

Instead of going outside when location is police car:
	say "'Moment mal' sagt der Polizeibeamte. 'Wo wollen Sie denn hin in diesem Zustand? Erzählen Sie mir erst mehr, damit ich ihnen helfen kann.'".

The furniture is female scenery in the police car. The printed name of the furniture is "Innenausstattung". Understand "Ausstattung" or "Innenausstattung" or "Inneres [n]" or "Innere [n]" as the furniture. The description of the furniture is "Im Augenblick gibt es wichtigeres, als die Innenausstattung zu betrachten.".

Instead of doing anything other than examining to the furniture:
	say "Es macht wenig Sinn, sich mit der Innenausstattung zu beschäftigen.".


Section - Display images after going to location

After going to the police car for the first time:
	show-left figure police;
	try looking.

After going to backyard:
	show-left figure backyard;
	try looking.

After going to left path:
	show-left figure leftpath;
	try looking.

After going to right path:
	show-left figure rightpath;
	try looking.
	
After going to dumping ground:
	show-left figure dumping;
	try looking.

After going to dying forest:
	show-left figure dyingforest;
	try looking.

After going to clearing:
	show-left figure clearing;
	try looking.

After going to hospital:
	show-left figure title;
	try looking.

Part - The actors

Section - Mariel

The nurse is a woman. The printed name of the nurse is "Krankenschwester". Understand  "Krankenschwester" or "Schwester" or "Mariel" or "Frau" as the nurse. The description of the nurse is "[The nurse] ist etwa Mitte vierzig, hat blonde Haare und trägt einen weißen Kittel. Was dir wirklich Sorgen macht sind ihre hart wirkenden Gesichtszüge und die Tatsache, dass sich ihre Haut langsam von ihrem Gesicht abzulösen scheint. Ihre Augen sind irgendwie leblos und starr.".

Rule for writing a paragraph about the nurse:
	say "[one of][The nurse] wartet ungeduldig, mit strengem Blick, dass du dich wieder hinlegst.[or][The nurse] tritt ungeduldig von einem Fuß auf den anderen.[or][The nurse] schaut dich mit durchdringendem Blick an.[or][The nurse] macht dir unmissverständliche Zeichen, dich wieder hinzulegen.[or]Streng deutet [the nurse] auf das Bett.[at random]".

The nurse-hair is plural-named part of the nurse. The printed name of the nurse-hair is "blond[^] Haare". Understand "blond" and "Haare" as the nurse-hair. the description of the nurse-hair is "Die blonden Haare sehen ungepflegt und verfilzt aus und hängen in langen Strähnen herunter.".

The nurse-face is neuter part of the nurse. The printed name of the nurse-face is "Gesicht[-s]". Understand "Gesicht" or "Gesichtszuege [p]" or "Haut [f]" or "Hautfetzen [p]" as the nurse-face. the description of the nurse-face is "An vielen Stellen ihres Gesichts lösen sich Hautfetzen ab. Ihr Lächeln ist eine Farce.".

The gown is a male thing, worn by the nurse. The printed name of the gown is "weiß[^] Kittel[-s]". Understand "weiss" and "Kittel" and "Flecken [p]" as the gown. the description of the gown is "Der Kittel war vielleicht mal weiß. Jetzt ist er eher grau vor Dreck und mit seltsamen dunklen Flecken überzogen.".

Instead of taking the gown:
	say "Vergiß es."

Section - Mariel conversation

Instead of answering the nurse that when the player is asked:
	say "'Ähm, also ... ich ...'[paragraph break]Die Krankenschwester unterbricht dich. 'Hallo, ich bin Mariel und ich werde mich um Sie kümmern.' sagt sie und ein unterkühltes Lächeln umspielt ihre Lippen. 'Legen Sie sich erstmal wieder hin und ich bin gleich bei Ihnen.'";
	change the printed name of the nurse to "Mariel";
	now the nurse is proper-named;
	now the player is not asked.

Instead of saying yes in the presence of the nurse when the player is asked:
		say "Ja, also ... ich ...'[paragraph break]Die Krankenschwester unterbricht dich. 'Hallo, ich bin Mariel und ich werde Ihnen gleich einen Tee bringen.' sagt sie und ein unterkühltes Lächeln umspielt ihre Lippen. 'Legen Sie sich erstmal wieder hin und ich bin gleich bei Ihnen.'";
	change the printed name of the nurse to "Mariel";
	now the nurse is proper-named;
	now the player is not asked.

Instead of saying no in the presence of the nurse when the player is asked:
		say "Nein, also ... ich ...'[paragraph break]Die Krankenschwester unterbricht dich. 'Hallo, ich bin Mariel und sie sollten auf jeden Fall viel trinken.' sagt sie und ein unterkühltes Lächeln umspielt ihre Lippen. 'Legen Sie sich erstmal wieder hin und ich bringe Ihnen gleich einen Tee.'";
	change the printed name of the nurse to "Mariel";
	now the nurse is proper-named;
	now the player is not asked.

Instead of answering the nurse that:
	say "[one of]'Ähm, also ... ich ...'[or]'Ich wollte nur ...'[or]'Vielleicht könnte ich ...'[or]'Meinen sie ich ...'[or]'Ach bitte lassen Sie ...[or]'Könnten sie bitte ...'[at random][line break][The nurse] unterbricht dich. [one of]'Legen sie sich einfach wieder hin, ich kümmere mich gleich um sie.'[or]'Sie sind doch ein braver Patient, oder?'[or]'Ich befehle ihnen, sich augenblicklich hinzulegen.'[or]'In ihrem Zustand herumzulaufen, ist unverantwortlich.'[or]'Ich hole etwas, das ihre Schmerzen lindert, aber legen sie sich erst wieder hin.'[at random]".

Instead of asking the nurse to try exiting:
	say "'Aber, aber' erwidert die Krankenschwester und schenkt dir einen mitleidigen Blick. 'Ich möchte Ihnen doch helfen, bitte legen Sie sich wieder hin'.".

Instead of asking the nurse to try going south:
	say "'Ich werde ja gehen' sagt [the nurse], 'aber vorher gehen Sie wieder in ihr Bett zurück.'".

Instead of asking the nurse to try attacking the player:
	say "'Genau das wollte ich tun' sagt [the nurse], 'aber vorher gehen Sie wieder in ihr Bett zurück.'".

Instead of asking the nurse to try attacking the nurse:
	say "'Ich bin schon tot' sagt [the nurse] und ein diabolisches Grinsen umspielt ihr Gesicht, 'machen Sie sich daher um mich keine Sorgen.'".

Instead of asking the nurse to try doing something:
	say "'Sie wird dir sicher nicht gehorchen.".

Instead of asking the nurse about when the player is asked:
	say "Die Krankenschwester unterbricht dich. 'Beantworten Sie mir doch bitte erst meine Frage.'";

Instead of asking the nurse about:
	say "Dir fehlen irgendwie die Worte. Vielleicht solltest du versuchen, bekannte Themen anzusprechen.".

Instead of asking the nurse about "brechstange/stange":
	if the nurse does not know about the crowbar:
		say "'Was für eine dumme Idee' sagt [the nurse]. 'Sie brauchen jetzt keine Brechstange, Sie müssen sich ausruhen.'";
		now the nurse knows about the crowbar;
	otherwise:
		annoy the nurse.

Instead of asking the nurse about "tuer/ausgang":
	if the nurse does not know about the white door:
		say "'Sie sollten die Türe immer geschlossen halten' sagt [the nurse]. 'Sonst bekommen Sie noch einen Zug und erkälten sich. Legen Sie sich wieder auf das Bett und ich bin gleich wieder bei Ihnen.'";
		now the nurse knows about the white door;
	otherwise:
		annoy the nurse.

Instead of asking the nurse about "bett/liege":
	if the nurse does not know about the bed:
		say "'Davon rede ich doch die ganze Zeit' sagt [the nurse]. 'Sie brauchen jetzt viel Ruhe.'";
		now the nurse knows about the bed;
	otherwise:
		annoy the nurse.

Instead of asking the nurse about "schrank":
	if the nurse does not know about the cabinet:
		say "'Der Schrank ist vollkommen unwichtig' sagt [the nurse]. 'Konzentrieren Sie sich jetzt auf ihre Genesung.'";
		now the nurse knows about the cabinet;
	otherwise:
		annoy the nurse.

Instead of asking the nurse about "fenster/gitter":
	if the nurse does not know about the window:
		say "'Das Gitter ist zu Ihrer eigenen Sicherheit da' sagt [the nurse]. 'Sie sehen ja selbst wie unruhig Sie sind. Begeben Sie sich sofort zurück ins Bett.'";
		now the nurse knows about the window;
	otherwise:
		annoy the nurse.

Instead of asking the nurse about "unfall/nacht/auto/splitter/strasse":
	if the nurse does not know about the accident:
		say "'Sie waren sehr leichtsinnig' sagt [the nurse]. 'Sie hatten viel Glück, Sie könnten jetzt tot sein'. Ein seltsames Lächeln umspielt ihre Lippen.";
		now the nurse knows about the accident;
	otherwise:
		annoy the nurse.

Instead of asking the nurse about "mich/mir/dich/dir": 
	if the nurse does not know about the player:
		say "'Gott sei Dank haben wir Sie gefunden' sagt [the nurse]. 'Sie hätten verbluten können dort draußen. Machen Sie sich keine Sorgen, ich sorge für Sie.'";
		now the nurse knows about the player;
	otherwise:
		annoy the nurse.

Instead of asking the nurse about "ihr/mariel":
	if the nurse does not know about the nurse:
		say "'Ich bin hier völlig unwichtig' sagt [the nurse]. 'Das wichtigste ist, dass Sie schnell wieder gesund werden.'";
		now the nurse knows about the nurse;
	otherwise:
		annoy the nurse.

Instead of asking the nurse about "zimmer/raum/ort/kammer/geruempel/ausgang":
	if the nurse does not know about the rubbish:
		say "'Der Raum ist nicht der schönste, ich weiß' sagt [the nurse]. 'Sie bekommen einen anderen, wenn die Belegung es zulässt.'";
		now the nurse knows about the rubbish;
	otherwise:
		annoy the nurse.

The church is a thing.

Instead of asking the nurse about "krankenhaus/hospital/gebaeude/klinik":
	if the nurse does not know about the church:
		say "'Wir sind eine sehr moderne Klinik mit einer enormen Erfolgsquote' sagt [the nurse]. 'Bei uns beträgt die Genesungsrate fast 100 Prozent. Sie sehen, bei uns sind Sie in besten Händen.'";
		now the nurse knows about the church;
	otherwise:
		annoy the nurse.

The i7 is a thing.

Instead of asking the nurse about "inform/gerx/inform7/if/i7":
	if the nurse does not know about the i7:
		say "'Davon habe ich bisher gar nichts gehört' sagt [the nurse]. 'Aber das macht nichts. Legen Sie sich einfach wieder hin.'";
		now the nurse knows about the i7;
	otherwise:
		annoy the nurse.

The pain is a thing.

Instead of asking the nurse about "spritze/infusion/injektion/fluessigkeit/medikament/medikamente/hilfe/schmerzen/schmerz/wunden/wunde/schnittwunden/schnittwunde/verletzung/verletzungen":
	if the nurse does not know about the pain:
		say "'Ja, ja' sagt [the nurse] ungeduldig. 'Ich hole ja gleich etwas; wenn Sie sich nur wieder brav hinlegen.'";
		now the nurse knows about the pain;
	otherwise:
		annoy the nurse.

Instead of asking the nurse about "haut/hautfetzen/gesicht":
	if the nurse does not know about the nurse-face:
		say "'Ach das' sagt [the nurse]. 'Das ist nur ein leichter Sonnenbrand. Nichts, worüber Sie sich Gedanken machen sollten.'";
		now the nurse knows about the nurse-face;
	otherwise:
		annoy the nurse.

Instead of asking the nurse about "zombie/zombies/untote":
	if the nurse does not know about the caretaker:
		say "'Ich weiß nicht genau, was Sie meinen' sagt [the nurse] ausweichend. 'Ich habe noch nie davon gehört.'";
		now the nurse knows about the caretaker;
	otherwise:
		annoy the nurse.

Interior is a thing.

Instead of asking the nurse about "korridor/flur/flure/korridore":
	if the nurse does not know about the interior:
		say "'Glauben Sie mir' sagt [the nurse]. 'Sie wollen nicht wissen, was da draußen so alles passiert. Es ist das Beste, Sie ruhen sich erstmal aus.'";
		now the nurse knows about the interior;
	otherwise:
		annoy the nurse.

The night is a thing.

Instead of asking the nurse about "dunkelheit/nacht":
	if the nurse does not know about the night:
		say "'Ich mag die Dunkelheit' sagt [the nurse]. 'Sie ist eine Wohltat, denn sie verdeckt die Ereignisse des Tages.'";
		now the nurse knows about the night;
	otherwise:
		annoy the nurse.

The weather is a thing.

Instead of asking the nurse about "wetter":
	if the nurse does not know about the weather:
		say "'Mit dem Wetter' sagt [the nurse]. 'können Sie sich morgen noch beschäftigen. Sie müssen jetzt schlafen.'";
		now the nurse knows about the weather;
	otherwise:
		annoy the nurse.

The doctor is a man.

Instead of asking the nurse about "arzt/aerzte/chefarzt":
	if the nurse does not know about the doctor:
		say "'Die Ärzte sind sehr beschäftigt heute' sagt [the nurse]. 'Wir hatten mehrere schwere Verkehrsunfälle in der Gegend. Sie können sicher sein, dass der Arzt morgen zu Ihnen kommt.'";
		now the nurse knows about the doctor;
	otherwise:
		annoy the nurse.

The eye is a thing.

Instead of asking the nurse about "augen/auge":
	if the nurse does not know about the eye:
		say "'Machen Sie sich keine Sorgen um mich' sagt [the nurse] mit einer monotonen Stimme, 'sondern eher um Sie selbst. Ruhen Sie sich aus.'";
		now the nurse knows about the eye;
	otherwise:
		annoy the nurse.

To annoy the nurse:
	say "[one of]'Da sehen Sie selbst, wie verwirrt Sie sind', merkt [the nurse] an, 'Dieses Thema hatten wir doch vorhin schon.'[or]'Sie wiederholen sich ständig', sagt [the nurse] genervt.[or]'Sie sind ja komplett durcheinander', merkt [the nurse] an, 'Wir haben doch bereits darüber gresprochen.'[or]'Sie brauchen dringend Ruhe', sagt [the nurse], 'Sie erzählen mir immer dasselbe.'[or]'Legen Sie sich doch bitte hin' sagt [the nurse], 'Ich hole gleich was für Sie.'[at random]".

Section - Policeman

The policeman is a man in the police car. The printed name of the policeman is "Polizeibeamt[^]". Understand  "Polizei" or "Polizeibeamter" or "Beamt" or "Beamter" or "Mann" or "Bulle" or "Cop" or "Polizist" or "kommissar" or "goerlich" as the policeman. The description of the policeman is "[The policeman] ist schätzungsweise Ende fünzig, mit braunen Haaren, einem Vollbart und einer dicken Zigarre im Mund. Er wirkt wie einer dieser typischen, sturen Beamten.".

Rule for writing a paragraph about the policeman:
	if the player is asked:
		now the policeman is mentioned;
	otherwise:
		say "[one of][The policeman] wartet ungeduldig auf Erklärungen.[or][The policeman] schaut dich mitleidig an.[or][The policeman] spielt nervös mit seinen Fingern.[or][The policeman] wirft dir einen skeptischen Blick zu.[or]Erwartungsvoll blickt dich [the policeman] an.[at random]".

The police-hair is plural-named part of the policeman. The printed name of the police-hair is "braun[^] Haare". Understand "braun" and "Haare" as the police-hair. the description of the police-hair is "Die braunen Haare sehen etwas ungepflegt aus.".

The police-face is neuter part of the policeman. The printed name of the police-face is "Gesicht[-s]". Understand "Gesicht" or "Gesichtszuege [p]" or "Haut [f]" or "Vollbart [m]" as the police-face. the description of the police-face is "Der Polizeibeamte trägt einen Vollbart, der ihn wahrscheinlich etwas älter macht, als er eigentlich ist.".

The cigar is a female thing, carried by the policeman. The printed name of the cigar is "dick[^] Zigarre". Understand "dick" and "Zigarre" and "Longfiller" and "Longfiller-Zigarre" as the cigar. the description of the cigar is "Eine von diesen dicken Longfiller-Zigarren, aus ganzen Blättern gerollt.".

The policeman is either convinced or suspicious. the policeman is suspicious.

The policeman has a number called suspiciousness. The suspiciousness of the policeman is 5.

Definition: the policeman is convinced if suspiciousness of the policeman is 0.

Section - Policeman conversation

Instead of answering the policeman that when the player is asked:
	say "'Ähm, also ... ich ...'[paragraph break]Der Polizeibeamte unterbricht dich. 'Guten Abend, ich bin Kommissar Görlich und ich werde mich um Sie kümmern.' sagt er und grinst dich an. 'Schnaufen sie erst mal durch, dann erzählen sie mir mal, was sie hier um diese Zeit machen.'";
	change the printed name of the policeman to "Kommissar Görlich";
	now the policeman is proper-named;
	now the player is not asked.

Instead of saying yes in the presence of the policeman when the player is asked:
	say "Ja, also ... ich ...'[paragraph break]Der Polizeibeamte unterbricht dich. 'Guten Abend, ich bin Kommissar Görlich und ich werde mich um Sie kümmern.' sagt er und grinst dich an. 'Schnaufen sie erst mal durch, dann erzählen sie mir mal, was sie hier um diese Zeit machen.'";
	change the printed name of the policeman to "Kommissar Görlich";
	now the policeman is proper-named;
	now the player is not asked.

Instead of saying no in the presence of the policeman when the player is asked:
	say "Nein, also ... ich ...'[paragraph break]Der Polizeibeamte unterbricht dich. 'Guten Abend, ich bin Kommissar Görlich und ich werde mich um Sie kümmern.' sagt er und grinst dich an. 'Schnaufen sie erst mal durch, dann erzählen sie mir mal, was sie hier um diese Zeit machen.'";
	change the printed name of the policeman to "Kommissar Görlich";
	now the policeman is proper-named;
	now the player is not asked.

Instead of answering the policeman that:
	say "[one of]'Ähm, also ... ich ...'[or]'Ich wollte nur ...'[or]'Vielleicht könnte ich ...'[or]'Meinen sie ich ...'[or]'Ach bitte lassen Sie ...[or]'Könnten sie bitte ...'[at random][line break][The policeman] unterbricht dich. [one of]'Sagen sie doch einfach frei heraus, was sie bedrückt.'[or]'Sie sollten wirklich zurück ins Krankenhaus; mit diesen Verletzungen ist nicht zu spaßen.'[or]'Was haben Sie denn auf dem Herzen?'[or]'Kann ich Ihnen irgendwie helfen?'[or]'In ihrem Zustand nachts auf der Straße herumzulaufen, ist unverantwortlich.'[at random]".

Instead of asking the policeman to try doing something:
	say "'Es ist wenig hilfreich, Polizisten Befehle erteilen zu wollen.".

Instead of asking the policeman about when the player is asked:
	say "Der Polizeibeamte unterbricht dich. 'Beantworten Sie mir doch bitte erst meine Frage.'";

Instead of asking the policeman about:
	say "Dir fehlen irgendwie die Worte. Vielleicht solltest du versuchen, bekannte Themen anzusprechen.".

Instead of asking the policeman about "brechstange/stange/brecheisen":
	if the policeman does not know about the crowbar:
		say "'Sie sind mit einer Brechstange aus der Klinik ausgebrochen?' sagt [the policeman]. 'Was für eine dumme Idee.'[paragraph break]Deine Antwort hat [den policeman] offenbar noch misstrauischer gemacht.";
		now the policeman knows about the crowbar;
		increase suspiciousness of the policeman by 2;
	otherwise:
		annoy the policeman.

Instead of asking the policeman about "bett/liege":
	if the policeman does not know about the bed:
		say "'Sie haben mit dem Bett die Tür blockiert, damit sie fliehen konnten?' sagt [the policeman]. 'Wie kommen Sie denn darauf?' Er kratzt sich nachdenklich am Nacken. [paragraph break]Deine Antwort hat [den policeman] offenbar noch misstrauischer gemacht.";
		now the policeman knows about the bed;
		increase suspiciousness of the policeman by 2;
	otherwise:
		annoy the policeman.

Instead of asking the policeman about "schrank":
	if the policeman does not know about the cabinet:
		say "'Sie haben diese Brechstange also aus dem Schrank gestohlen?' sagt [the policeman]. 'Das ist nicht unwesentlich für das Protokoll und trägt nicht gerade zu Ihrer Entlastung bei.'[paragraph break]Deine Antwort hat [den policeman] offenbar noch misstrauischer gemacht.";
		now the policeman knows about the cabinet;
		increase suspiciousness of the policeman by 2;
	otherwise:
		annoy the policeman.

Instead of asking the policeman about "fenster/gitter":
	if the policeman does not know about the window:
		say "'Die Fenster waren vergittert?' sagt [the policeman]. 'Das ist in der Tat sehr seltsam und klingt mehr nach geschlossener Anstalt als nach Klinik.'[paragraph break]Deine Antwort hat [den policeman] offenbar weniger misstrauisch gemacht.";
		now the policeman knows about the window;
		decrease suspiciousness of the policeman by 2;
	otherwise:
		annoy the policeman.

The accident is a thing.

Instead of asking the policeman about "unfall/nacht/auto/splitter/strasse":
	if the policeman does not know about the accident:
		say "'Ich notiere: Sie hatten einen schweren Verkehrsunfall letzte Nacht.' sagt [the policeman]. 'Darüber finde ich sicher etwas in den Akten.'[paragraph break]Deine Antwort hat [den policeman] offenbar weniger misstrauisch gemacht.";
		now the policeman knows about the accident;
		decrease suspiciousness of the policeman by 1;
	otherwise:
		annoy the policeman.

Instead of asking the policeman about "mich/mir/dich/dir", say "'Jetzt erzählen Sie erst mal in Ruhe was passiert ist' sagt [the policeman]. 'Dann sehen wir weiter.'"

Instead of asking the policeman about "ihm/polizei/beamter/polizeibeamter/polizist/bulle/cop", say "'Ich bin jetzt erst mal unwichtig' sagt [the policeman]. 'Erzählen Sie mir lieber etwas von Ihnen.'"

Instead of asking the policeman about "mariel/krankenschwester/schwester/personal", say "'Die Krankenschwester war seltsam?' sagt [the policeman]. 'Na, ja, manchmal kommt es vor, dass man mit dem Personal nicht so gut auskommt. Kein Grund zur Beunruhigung.'[paragraph break]Deine Antwort hat [den policeman] offenbar nicht überzeugt.".

Instead of asking the policeman about "zimmer/raum/ort/kammer/gerümpel/ausgang", say "'Sie waren in einer Abstellkammer untergebracht?' sagt [the policeman]. 'Das war sicher ein Mißverständnis.'[paragraph break]Deine Antwort hat [den policeman] offenbar nicht überzeugt.".

Instead of asking the policeman about "krankenhaus/hospital/gebaeude/klinik/kathedrale/kirche":
	if the policeman does not know about the church:
		say "'Sie meinen das alte Hospital.' sagt [the policeman]. 'Sie müssen sich irren. Soweit ich weiß, ist das bereits vor Jahren geschlossen worden.'[paragraph break]Deine Antwort hat [den policeman] offenbar nicht überzeugt.";
		now the policeman knows about the church;
	otherwise:
		annoy the policeman.

Instead of asking the policeman about "inform/gerx/inform7/if/i7":
	if the policeman does not know about the i7:
		say "'Mein Gott, Sie sind ja ein echter Nerd.' sagt [the policeman]. 'Was machen Sie sonst, außer an ihrem Computer zu sitzen?'";
		now the policeman knows about the i7;
	otherwise:
		annoy the policeman.

Instead of asking the policeman about "schmerz/wunden/wunde/schnittwunden/schnittwunde/verletzung/verletzungen":
	if the policeman does not know about the player:
		say "'Oh mein Gott, Sie haben ja zahlreiche Verletzungen.' sagt [the policeman]. 'Ich bringe Sie sofort in die Klinik zurück.' Er sieht irgendwie blass aus.[paragraph break]Deine Antwort hat [den policeman] offenbar noch misstrauischer gemacht.";
		now the policeman knows about the player;
		increase suspiciousness of the policeman by 1;
	otherwise:
		annoy the policeman.

Instead of asking the policeman about "haut/hautfetzen/gesicht/zombie/augen/auge/zombies/untote", say "'Sie haben dort Leute mit Sonnenbrand gesehen?' sagt [the policeman]. 'Nicht verwunderlich, bei dieser Hitze tagsüber.'".

Instead of asking the policeman about "korridor/flur/flure/korridore":
	if the policeman does not know about the interior:
		say "'Ja ,klar.' sagt [the policeman]. 'Sie sprechen allem Anschein nach von der alten Kirche, die um die Jahrhundertwende zum Hospital ausgebaut wurde. Es ist vor Jahren schon geschlossen worden. Was Sie da erzählen, klingt in der Tat merkwürdig.'[paragraph break]Deine Antwort hat [den policeman] offenbar weniger misstrauisch gemacht.";
		now the policeman knows about the interior;
		decrease suspiciousness of the policeman by 1;
	otherwise:
		annoy the policeman.

Instead of asking the policeman about "schacht/bewegung/tiefe/dunkelheit":
	if the policeman does not know about the shaft:
		say "'Ein tiefer dunkler Schacht sagen Sie?' sagt [the policeman]. 'Und da hat sich unten irgendetwas bewegt? Das klingt ja fast nach einem Horrorfilm. Wie auch immer, die Polizei wird das aufdecken.'[paragraph break]Deine Antwort hat [den policeman] offenbar weniger misstrauisch gemacht.";
		now the policeman knows about the shaft;
		decrease suspiciousness of the policeman by 1;
	otherwise:
		annoy the policeman.

Instead of asking the policeman about "falltuer/falltuere/deckel":
	if the policeman does not know about the wooden trapdoor:
		say "Warum soll eine Klinik eine Falltüre haben?' sagt [the policeman]. 'Das klingt ja ziemlich verdächtig. Da genehmige ich mir erst mal einen Schluck.' Er nimmt einen Flachmann heraus, trinkt davon und steckt ihn wieder ein.[paragraph break]Deine Antwort hat [den policeman] offenbar weniger misstrauisch gemacht.";
		now the policeman knows about the wooden trapdoor;
		decrease suspiciousness of the policeman by 1;
	otherwise:
		annoy the policeman.

Instead of asking the policeman about "baum/blaetter/baeume/laub":
	if the policeman does not know about the wooden trapdoor:
		say "'Es waren da tatsächlich abgestorbene Bäume im Hochsommer zu finden?' sagt [the policeman]. 'Sehr verdächtig.'[paragraph break]Deine Antwort hat [den policeman] offenbar weniger misstrauisch gemacht.";
		now the policeman knows about the wooden trapdoor;
		decrease suspiciousness of the policeman by 1;
	otherwise:
		annoy the policeman.

Instead of asking the policeman about "dunkelheit/nacht", say "'Sind sie wahnsinnig, bei dieser Dunkelheit hier draußen herumzuirren?' sagt [the policeman]. 'Ich bringe Sie am besten zurück in die Klinik.'".

Instead of asking the policeman about "wetter", say "'Jetzt kommen Sie schon' sagt [the policeman]. 'Was soll der smalltalk über's Wetter? Sagen Sie schon, was Sie beschäftigt.'".

Instead of asking the policeman about "koloss/mann/huene":
	if the policeman does not know about the caretaker:
		say "'Sie haben einen Mann gesehen, der zwei Meter achtzig groß ist?' sagt [the policeman]. 'Das ist in der Tat zeimlich ungewöhnlich.' Er kratzt sich nachdenklich am Kopf. 'Die Polizei wird Ihre Aussage prüfen.'[paragraph break]Deine Antwort hat [den policeman] offenbar weniger misstrauisch gemacht.";
		now the policeman knows about the caretaker;
		decrease suspiciousness of the policeman by 1;
	otherwise:
		annoy the policeman.		

Instead of asking the policeman about "infusion/spritze/tod/injektion/dosis/fluessigkeit/medikament/medikamente/hilfe/schmerzen":
	if the policeman does not know about the nurse:
		say "'Sie sollten umgebracht werden?' sagt [the policeman]. 'In einem Hospital? Das ist eine schwere Anschuldigung, das ist Ihnen doch klar, oder?'[paragraph break]Deine Antwort hat [den policeman] offenbar weniger misstrauisch gemacht.";
		now the policeman knows about the nurse;
		decrease suspiciousness of the policeman by 2;
	otherwise:
		annoy the policeman.		

Instead of asking the policeman about "flucht/verfolgung", say "'Sie leiden ja an Verfolgungswahn, sind sie gar aus der Klinik geflohen?' sagt [the policeman]. 'Ich bringe Sie am besten dorthin zurück.'".

To annoy the policeman:
	say "[one of]'Das haben Sie bereits gesagt' merkt [the policeman] an.[or]'Sie wiederholen sich ständig' sagt [the policeman] genervt.[or]'Ja,ja, das mag schon sein' sagt [the policeman], 'aber es ist noch nicht bewiesen.'[or]'Haben Sie was genommen?' fragt [the policeman] 'Sie erzählen mir immer dasselbe.'[or]'Sind sie ok?' fragt [the policeman] 'Ich finde, Sie sollten mal das Thema wechseln.'[at random]";
	if a random chance of 1 in 2 succeeds:
		increase suspiciousness of the policeman by 1.

Section - The caretaker

The caretaker is a man in the dumping ground. The printed name of the caretaker is "Koloss[-es]". understand "Riese" or "Koloss" or "Mann" or "Ungeheuer [n]" or "Huene" as the caretaker. the description of the caretaker is "Der Mann ist ein Hüne von gut zwei Meter achtzig Größe; breit und muskulös, wie ein Berg steht er vor dir. Große Teile seiner Haut fehlen. Am beängstigsten jedoch ist sein Gesicht: offenes Fleisch, Knochen und leere, weiße, starre Augen.".

Distance is a number that varies. Distance is 3.

After going somewhere when the location is in outdoors:
	change distance to 3;
	if hunting is happening:
		move the caretaker to the location;
	continue the action.

The small tree is a male thing. The printed name of the small tree is "klein[^] Baum[-es]". Understand "klein" and "baum" as the small tree. the description of the small tree is "Ein kleiner Baum mit ungefähr zwei Meter Kronendurchmesser; der Koloss wiegt ihn in der Hand hin und her.".

Does the player mean examining the small tree:
	it is likely.

Instead of attacking the caretaker for the first time:
	if the player is holder of the crowbar:
		say "Mutig trittst du dem Koloss in den Weg. Angriffslustig schwenkst du die Brechstange in der Hand. Der Mann zeigt sich vollkommen unbeeindruckt und reißt einen kleinen Baum aus dem Boden aus. Er trifft dich mit voller Wucht und schleudert dich gut drei Meter nach hinten.";
		now the small tree is carried by the caretaker;
	otherwise:
		say "Mit bloßen Händen hast du gegen diesen Koloss nicht den Hauch einer Chance.".

Instead of attacking or pushing or pulling or turning the caretaker:
	if the player is holder of the crowbar:
		say "Die pure Verweiflung treibt dich erneut zu einem Angriff; du triffst den Koloss am Arm, aber der Angriff zeigt nicht die geringste Wirkung. Noch einmal holt der Mann mit dem Baum aus und trifft dich am Kopf. Du bist einen Augenblick lang bewusstlos. Der Mann schleift dich mit sich, langsam und unaufhaltsam. Als du wieder zu dir kommst, schleppt er dich die Stufen zum Vordereingang des schwarzen Gebäudes hinauf.";
		end the story finally saying "Du bist verloren.";
	otherwise:
		say "Mit bloßen Händen hast du gegen diesen Koloss nicht den Hauch einer Chance.".

The care-face is neuter part of the caretaker. The printed name of the care-face is "Gesicht[-s]". Understand "gesicht" or "gesichtszuege [p]" or "haut [f]" or "hautfetzen [p]" or "Augen [p]" as the care-face. the description of the care-face is "Er leidet offenbar an einer Art Krankheit im Endstadium. Sein Gesicht besteht nur noch zu einem kleinen Teil aus Haut, der Rest ist offenliegendes Fleisch, Knochen und Muskeln; ein Anschauungsobjekt der menschlichen Anantomie.".

Section - Caretaker conversation

Instead of answering the caretaker that for the first time:
	say "'Du hörst nur ein bedrohliches Schnaufen.".

Instead of answering the caretaker that:
	say "Du hörst ein Röcheln.".
	
Instead of asking the caretaker about:
	say "Irgendetwas sagt dir: jetzt ist nicht die Zeit, zu reden.".

Instead of asking the caretaker to try doing something:
	say "'Jeglicher Versuch ist zwecklos.".

Part - The scenes

Section - Scene investigating

Investigating is a scene.

Investigating begins when the player is in the police car.

Investigating ends sadly when the the player is in the police car for the 15th turn.

Investigating ends happy when the policeman is convinced.

When investigating begins:
	say "'Setzen sie sich erstmal hin und beruhigen sie sich' sagt der Polizeibeamte. 'Haben Sie sich verletzt?'";
	now the player is asked;
	if the policeman can see the crowbar:
		say "[line break]Was haben wir denn da?' sagt der Polizeibeamte. 'Eine Brechstange.' Er blickt dich argwöhnisch an. Das hat sein offensichtliches Misstrauen nicht gerade verringert.";
		increase suspiciousness of the policeman by 4.

When investigating ends happy:
	say "'Ich werde das in Ruhe näher untersuchen' sagt [the policeman]. 'Ich fahre sie jetzt erst mal nach Hause.'";
	end the story finally saying "Du bist entkommen.";

When investigating ends sadly:
	say "Bevor du noch irgendetwas tun kannst, tauchen seltsame Gestalten aus der Finsternis auf. Sie haben den Polizeiwagen umstellt, es gibt kein Entkommen mehr. Mit einem Ausdruck des Entsetzens in seinem Gesicht, zieht der Polizist seine Pistole. Er gibt einen Warnschuss ab, aber die Gestalten lassen sich nicht berirren, langsam wanken sie weiter auf den Wagen zu. Der Polizist schießt wahllos in die Menge, aber die Getroffenen lassen sich von Pistolenkugeln nicht aufhalten. Sie öffnen die Autotür zerren dich nach draußen.";
	end the story finally saying "Du bist verloren.".

Section - Hunting

Hunting is a scene.

Hunting begins when the caretaker can see the player.

Rule for writing a paragraph about the caretaker:
	now the caretaker is mentioned.

When hunting begins:
	now the player knows about the caretaker.

Every turn during hunting:
	decrease distance by 1;
	if location is the dumping ground:
		if distance is 2:
			say "Der Mann bemerkt dich sofort und dreht langsam den Kopf; er rümpft seine Nase, als wenn er dich riechen könnte und beginnt langsam, mit wankenden Schritten auf dich zuzugehen.";
		if distance is 1:
			say "Der Mann wankt weiter langsam aber stetig auf dich zu; Panik ergreift dich. Sein Blick ist leer und kalt. Noch wenige Schritte und er erreicht dich.";
	if location is the right path:
		if distance is 2:
			say "Der Mann wankt weiter hinter dir her; er bewegt sich langsam den Weg entlang, direkt in deine Richtung. Wieder dreht er den Kopf hin und her und rümpft seine Nase, wie ein Tier, das deine Witterung aufnimmt.";
		if distance is 1:
			say "Der Mann kommt dir bedrohlich nahe; deine Beine beginnen zu zittern. Die Augen des Mannes sind leer und weiß. Noch wenige Schritte und er erreicht dich.";
	if location is the backyard:
		if distance is 2:
			say "Der Mann hält kurz inne; er gibt ein kurzes Grunzen von sich. Wieder dreht der Mann sich in deine Richtung. Es scheint kein Entkommen zu geben. Langsam, unablässig kommt er mit wankenden Schritten auf dich zu.";
		if distance is 1:
			say "Wiederum trennen dich nur noch wenige Schritte von dem Mann; deine Beine fühlen sich müde und schwer an, als wollen sie ihren Dienst versagen.";
	if location is the left path:
		if distance is 2:
			say "Der Mann wankt weiter hinter dir her; er bewegt sich langsam den Weg entlang. Wieder dreht er den Kopf hin und her und hält einen Moment inne; dann bewegt er sich zielstrebig weiter auf dich zu.";
		if distance is 1:
			say "Wieder kommt der Mann dir bedrohlich nahe; dein Herz pocht wie wild. Die Wangen des Mannes sind eingefallen und er ist blaß im Gesicht. Noch wenige Schritte und er erreicht dich.";
	if location is the dying forest:
		if distance is 2:
			say "Raschelnd bewegt sich der Mann über den laubbedeckten Boden. Mit wankenden Schritten kommt er auf dich zu, sein Gesicht ist vor Wut verzerrt.";
		if distance is 1:
			say "Der Mann scheint übermenschliche Kräfte zu haben, Panik ergeift dich. Noch wenige Schritte und er erreicht dich.";
	if location is the clearing:
		if distance is 2:
			say "Jetzt gibt es keinen Ausweg mehr. Der Koloss kommt unaufhaltsam auf dich zu und diese Lichtung ist von undurchdringlichen dicht stehenden Bäumen umgeben; kein Weg vor und keiner zurück.";
		if distance is 1:
			if the leaves are not on the wooden trapdoor:
				say "Der Mann wankt weiter auf dich zu; er bemerkt den Holzdeckel und weicht ihm geschickt aus. In Panik hältst du deine Hände schützend vor deine Augen. Er hat dich fast erreicht.";
			otherwise:
				say "Der Mann wankt weiter auf dich zu; er tritt jetzt auf den laubbedeckten Holzdeckel und bleibt für einen Moment lang kurz stehen.";
				move the caretaker to the wooden trapdoor.

Section - Blocking directions during hunting

Instead of going south from dumping ground when hunting is happening:
	say "Du siehst keinen Weg, wie du an dem Koloss von Mann vorbei kommen kannst.".

Instead of going south from right path when hunting is happening:
	say "Der Pfad ist viel zu eng, um an dem Mann vorbeizukommen.".

Instead of going east from backyard when hunting is happening:
	say "Du siehst keinen Weg, wie du an dem Koloss von Mann vorbei kommen kannst.".

Instead of going east from left path when hunting is happening:
	say "Der Pfad ist viel zu eng, um an dem Mann vorbeizukommen.".
	
Instead of going north from dying forest when hunting is happening:
	say "Du siehst keinen Weg, wie du an dem Koloss von Mann vorbei kommen kannst.".

Instead of going north from clearing when hunting is happening:
	say "Du siehst keinen Weg, wie du an dem Koloss von Mann vorbei kommen kannst.".

Hunting ends sadly when distance is 0.

Hunting ends happily when the caretaker is in the pit.

When hunting ends happily:
	say "Du atmest tief durch. Das war wirklich knapp.".

When hunting ends sadly:
	say "Der Mann packt dich am Arm. Er schleift dich mit sich, langsam, aber unaufhaltsam. Du wehrst dich, du schlägst wie wild mit den Fäusten auf ihn ein; vergeblich. Du siehst noch, wie er sich die Stufen zum Vordereingang des schwarzen Gebäudes hinaufschleppt, als dich ein harter Schlag ins Gesicht trifft und du besinnungslos wirst.";
	end the story finally saying "Du bist verloren.".

Section - Scene blocking

Blocking is a scene. Blocking begins when the player can see the nurse.

When blocking begins:
	say "Gerade als du den Raum verlassen willst, steht urplötzlich eine Krankenschwester in der Tür. 'Aber, aber, mein Bester. Wer wird denn schon aufstehen?', sagt sie mit einem kühlen Unterton und betritt das Zimmer. 'Soll ich Ihnen etwas zu trinken bringen?'";
	now the player is asked.

Blocking ends naturally when the player is on the bed.

Blocking ends abruptly when time of day is after 09:23 AM.

Blocking ends instantly when the nurse can see the crowbar.

When blocking ends naturally:
	say "'So ist es brav, sie brauchen jetzt viel Ruhe' sagt [the nurse] und schließt die Tür. Mit einem leisen Klicken stirbt deine letzte Hoffnung auf Rettung.";
	now the white door is closed;
	now the white door is locked;
	move the nurse to the corridor.

When blocking ends abruptly:
	say "Mit einer kraftvollen Bewegung stößt dich [the nurse] auf das Bett. 'So ist es brav, sie brauchen jetzt viel Ruhe' sagt sie und schließt die Tür. Mit einem leisen Klicken stirbt deine letzte Hoffnung auf Rettung.";
	now the white door is closed;
	now the white door is locked;
	move the nurse to the corridor;
	move the player to the bed.

When blocking ends instantly:
	say "'Was haben wir denn da?' schreit [the nurse]. Mit einer kraftvollen Bewegung reißt sie dir die Brechstange aus der Hand und stößt dich auf das Bett. 'So ist es brav, sie brauchen jetzt viel Ruhe' sagt sie und schließt die Tür. Mit einem leisen Klicken stirbt deine letzte Hoffnung auf Rettung.";
	now the white door is closed;
	now the white door is locked;
	move the nurse to the corridor;
	move the crowbar to the station-cabinet;
	now the station-cabinet is closed;
	move the player to the bed.

Section - Scene infusing

Infusing is a scene. Infusing begins when the time of day is 09:30 AM and location is hospital or location is west end or location is ventilation system.

When infusing begins:
	if location is not hospital:
		if bed is lengthways:
			say "Aus deinem Zimmer kommen seltsame Geräusche.";
		otherwise:
			show-left figure title;
			say "[if blocking has happened][Der nurse][otherwise][A nurse][end if] steckt ihren Kopf durch das Loch in der Kassettendecke. 'Aha, da sind sie' sagt sie und zerrt dich an deinen Füßen mit eiserner Kraft zurück nach unten auf das Bett.[paragraph break]";
			kill the player;
		stop the action;
	if the white door is locked:
		say "Mit einem Klicken wird die Tür wieder aufgeschlossen.[paragraph break]";
	if the bed is crossways:
		if blocking has happened:
			say "[The nurse] betritt die Kammer wieder.[paragraph break]";
		otherwise:
			say "[A nurse] öffnet die Tür und betritt die Kammer. Vor dir steht eine Frau, etwa Mitte vierzig, mit blonden Haaren und einem weißen Kittel. Was dir wirklich Sorgen macht sind ihre hart wirkenden Gesichtszüge und die Tatsache, dass sich ihre Haut anscheinend langsam von ihrem Gesicht ablöst.[paragraph break]";
		kill the player;
	otherwise:
		say "Jemand versucht, die Tür zu öffnen, aber das Bett blockiert die Tür.".

Every turn during infusing:
	if the bed is crossways:
		if blocking has happened:
			say "[The nurse] betritt die Kammer wieder.[paragraph break]";
		otherwise:
			say "[A nurse] öffnet die Tür und betritt die Kammer. Vor dir steht eine Frau, etwa Mitte vierzig, mit blonden Haaren und einem weißen Kittel. Was dir wirklich Sorgen macht sind ihre hart wirkenden Gesichtszüge und die Tatsache, dass sich ihre Haut anscheinend langsam von ihrem Gesicht ablöst.[paragraph break]";
		kill the player;
	otherwise:	
		if location is hospital:
			say "[one of]Jemand rüttelt heftig an der Tür.[or]Du hörst ein lautes Fluchen von draußen.[or]Ein ungeduldiges Klopfen hallt durch die Kammer.[or]Die Tür erbebt unter heftigen Schlägen.[at random]".

Section - To kill the player

To kill the player:
	say "Sie hält eine Spritze mit einer grünen Flüssigkeit in der Hand. Du willst dich wehren, aber eisige Finger halten dich mit einem unmenschlich starken Griff fest. Du spürst einen brennenden Schmerz, als sich die grüne Flüssigkeit langsam in deinen Venen ausbreitet.";
	end the story finally saying "Du bist verloren.".

Section - Go To

Going by name is an action applying to one thing. 

Understand "geh zu [any room]" or "geh nach [any room]" or "geh in [any room]" as going by name. 

Check going by name: 
	if the noun is the location, say "Aber du bist schon hier." instead; 
	if the noun is not adjacent and the noun is unvisited, say "Dieses Hauptwort macht in diesem Zusammenhang keinen Sinn." instead;
	if the noun is not adjacent and the noun is visited, say "[The noun] ist kein Ort, den du von hier erreichen kannst." instead.

Carry out going by name: 
    let aim be the best route from the location to the noun, using doors; 
    if aim is not a direction, say "Du hast keine Ahnung, wie du von hier dorthin gelangen kannst." instead; 
    try going aim.

Section - Despairing

Despairing is an action applying to nothing.
understand "panik" or "verzweifel" or "kapitulier" or "verzweifl" as despairing.

carry out despairing:
	say "Du kämpst gegen die Panik an - verzweifeln bringt dich sicher nicht weiter."

Section - Cringing

Cringing is an action applying to nothing.
understand "erschauder" or "schauder" or "frier" as cringing.

carry out cringing:
	say "Es läuft dir eiskalt den Rücken herunter."

Section - Dancing

Dancing is an action applying to nothing.
understand "tanz" as dancing.

Instead of turning yourself:
	try dancing.

carry out dancing:
	if location is the hospital or location is the police car:
		say "Zum tanzen ist es hier viel zu eng.";
		stop the action;
	say "Du tanzt ein paar Schritte - nicht gerade der Hit, aber es hilft gegen die beklemmende Angst."

Section - Praying

Praying is an action applying to nothing.
understand "bet" or "bet zu gott" or "bete zu gott" or "bete" as praying.

carry out praying:
	say "Du schickst ein Stoßgebet zum Himmel."

Section - Crying

Crying is an action applying to nothing.
understand "heul" or "heul doch" or "wein" as crying.

carry out crying:
	say "Weinen ist jetzt wenig hilfreich."

Section - Chatting

Chatting is an action applying to nothing.
understand "cool" or "geil" or "heftig" or "dumm" or "bloed" or "peinlich" or "schrecklich" or "scheußlich" or "gruselig" or "grausig" or "abartig" or "daemlich" or "nett" or "schoen" or "prima" or "na prima" or "klasse" or "na klasse" or "spitze" or "gut" or "schlecht" or "wahnsinn" or "endgeil" as chatting.

carry out chatting:
	say "Yep."

Section - Thinking

Instead of thinking:
	say "Hier [if location is indoors]im Raum [end if]sind [a list of visible other things with nominative].[paragraph break]";
	if the player can see the crowbar and location is hospital:
		say "Du bist dir sicher, dass die Brechstange dir helfen wird, hier heraus zu kommen.";
		stop the action;
	if  location is hospital:
		say "Verzweifelt siehst du dich um. Wenn du nur dieses Fenster öffnen könntest.";
	if location is outdoors and the caretaker is not in the pit:
		say "Wenn du diesen Koloss von Mann nur irgendwie unschädlich machen könntest.";
		stop the action;
	if location is outdoors:
		say "Du musst schnellstens von hier weg.";
	if location is police car:
		say "Wenn du diesen sturen Beamten nur überzeugen könntest.".

Section - Advising

Advising first is a truth state that varies. Advising first is true.

Advising second is a truth state that varies. Advising second is true.

Advisor is a device. Advisor is switched off.

The advise list is a list of things that varies.

Advising is an action applying to nothing.
understand "rat" or "ratgeber" or "helfer" as Advising.

After opening the something (called the prison):
	if advisor is switched on:
		if something other (called the invader) is in the cabinet:
			add the invader to the advise list;
	continue the action.

carry out Advising:
	if advisor is switched off:
		if location is not hospital:
			say "Diese Option ist leider nur am Anfang dieser Geschichte verfügbar.";
			stop the action;
		now advisor is switched on;
		if advising first is true:
			say "(Ratgeber Modus ist aktiviert)[italic type][paragraph break]Hallo, ich bin eine Stimme im Hintergrund, die dir Hilfestellung geben wird. Abhängig von den gerade sichtbaren Gegenständen an deinem Aufenthaltsort werde ich dir im Zufallsmodus eine sinnvolle Aktion damit vorschlagen. Ich leiste keinerlei Hilfestellung zur Lösung der Rätsel, sondern erteile nur einen Rat für die Interaktion mit der Spielwelt. Hast du alles probiert, was ich vorgeschlagen habe, erhältst du mit WEITER oder OK weiteren Rat von mir. Du kannst mich mit RATGEBER wieder ausschalten[roman type].";
			now advising first is false;
		otherwise:
			say "(Ratgeber Modus ist aktiviert)[italic type][paragraph break]Hallo, da bin ich wieder. Abhängig von den gerade sichtbaren Gegenständen an deinem Aufenthaltsort werde ich dir im Zufallsmodus eine sinnvolle Aktion damit vorschlagen. Wenn ich dir etwas sagen soll, dann gib WEITER oder OK ein. Du kannst mich mit RATGEBER wieder ausschalten[roman type].";		
		change advise list to the list of other touchable things;
		try advancing;
		stop the action;
	otherwise:
		now advisor is switched off;
		say "(Ratgeber Modus ist deaktiviert)[italic type][paragraph break]Ich wünsche dir viel Erfolg im weiteren Spielverlauf. Bis bald[roman type].".

Section - Advancing

Else to say is a truth state that varies. else to say is false.

Advancing is an action applying to nothing.
understand "weiter" or "ok" as Advancing.

Check Advancing:
	if advisor is switched off:
		say "Diese Option steht leider nur im Ratgeber Modus zur Verfügung. Willst du diesen Modus einschalten gibt RATGEBER ein."

Carry out Advancing:
	repeat with item running through advise list:
		if item is not touchable:
			remove item from advise list;
		if the player is carrying the item:
			remove item from advise list;
	if number of entries in advise list is 0:
		say "[italic type]Ich habe dir Interaktionsmöglichkeiten für alle hier sichtbaren Gegenstände gezeigt. Eventuell kannst du diesen Ort in eine angegebene Himmelsrichtung verlassen. Dazu gib z.B. einfach SÜDEN oder kurz S ein. Außerdem versteht das Spiel VERLASSE (den Ort) oder auch GEHE ZU (einem Ort).[paragraph break]Wenn Du mich wieder benötigst, gib nochmal RAT oder RATGEBER ein[roman type].";
		now advisor is switched off;
		stop the action;
	if the yourself does not know about the yourself:
		say "[italic type]In Ordnung, beginnen wir mit ein paar Grundlagen: du bist im Moment im [location of the player]. Dein Aufenthaltsort ist immer in der schwarzen Statuszeile oben links ablesbar, außerdem wird rechts die Anzahl deiner verbrauchten Spielzüge angezeigt. Die Beschreibung deines Aufenthaltsortes kannst du jedezeit mit SCHAU oder L abrufen.[paragraph break]Wenn du BETRACHTE DICH oder kurz B DICH eingibst, erhältst du eine entsprechende Information, mit INVENTAR oder kurz I siehst du, welche Gegenstände du im Augenblick bei dir trägst.[paragraph break]Im Übrigen können bei allen Befehlen die Artikel und die Adjektive problemlos weggelassen werden[roman type].";
		now the yourself knows about the yourself;
		stop the action;
	if the player is on something (called the prison):
		say "[italic type]Du befindest dich zur Zeit auf [dem prison]. Du könntest [one of]vielleicht[or]mal[or]eventuell[or]einfach[or]jetzt[at random] versuchen, [den prison] zu verlassen. Das geht z.B. mit RAUS oder RUNTER oder STEH AUF[roman type].";
		if prison is listed in advise list:
			remove the prison from the advise list;
		stop the action;
	if the player is in something (called the prison):
		say "[italic type]Du befindest dich zur Zeit in [dem prison]. Du könntest [one of]vielleicht[or]mal[or]eventuell[or]einfach[or]jetzt[at random] versuchen, [den prison] zu verlassen. Das geht z.B. mit RAUS[roman type].";
		if prison is listed in advise list:
			remove the prison from the advise list;
		stop the action;
	sort advise list in random order;
	let item be entry 1 of the advise list;
	remove item from advise list;
	If a person (called the owner) is holder of item:
		If item is part of the owner:
			say "[italic type][Der item] [if item is plural-named]wären[otherwise]wäre[end if] auf jeden Fall einen Bilck wert. Betrachte [den item] genauer mit B ([den item])[roman type].";
			stop the action;
		If item is worn by the owner:
			say "[italic type][Der owner] trägt [den item]. Betrachte [den item] genauer mit B ([den item])[roman type].";
			stop the action;
		If item is carried by the owner:
			say "[italic type][Der owner] trägt [den item] bei sich. Betrachte [den item] genauer mit B ([den item])[roman type].";
			stop the action;			
	otherwise if advising second is true:	
		say "[italic type]Hier [if item is plural-named]befinden[otherwise]befindet[end if] sich [ein item]. Betrachte [den item] genauer mit B ([den item]). Dieser Befehl sollte auch bei allen weiteren Gegenständen oder Personen immer der erste sein, den du eintippst[roman type].";
		now else to say is true;
		now advising second is false;
	if item is locked door:
		say "[Der item] [ist] verschlossen. Du könntest [one of]vielleicht[or]mal[or]eventuell[or]einfach[or]jetzt[at random] versuchen, [den item] aufzuschließen, sofern du einen passenden Schlüssel dabei hast. Das geht z.B. mit SCHLIESSE ([den item]) mit (etwas) auf[roman type].";
		stop the action;
	if item is closed door:
		if else to say is true:
			say "[line break]";
			now else to say is false;
		say "[italic type]Fenster und Türen sind meist zum Öffnen da. Das Spiel versteht z.B. ÖFFNE ([den item])[roman type].";
		stop the action;
	if item is a closed container:
		if else to say is true:
			say "[line break]";
			now else to say is false;
		say "[italic type][Der item] [ist] geschlossen [if item is plural-named]könnnen[otherwise]kann[end if] eventuell geöffnet werden. Versuch einfach mal ÖFFNE ([den item])[roman type].";
		stop the action;
	if item is enterable supporter:
		if else to say is true:
			say "[line break]";
			now else to say is false;
		say "[italic type]Du könntest [one of]vielleicht[or]mal[or]eventuell[or]einfach[or]jetzt[at random] versuchen, auf [den item] zu gelangen. Das Spiel versteht z.B. STEIG auf ([den item])[roman type].";
		stop the action;
	if item is supporter:
		if else to say is true:
			say "[line break]";
			now else to say is false;
		say "[italic type]Du könntest [one of]vielleicht[or]mal[or]eventuell[or]einfach[or]jetzt[at random] versuchen, etwas auf [den item] zu legen. Dafür gibt es den Befehl LEGE (etwas) AUF ([den item])[roman type].";
		stop the action;
	if item is enterable container:
		if else to say is true:
			say "[line break]";
			now else to say is false;
		say "[italic type]Du könntest [one of]vielleicht[or]mal[or]eventuell[or]einfach[or]jetzt[at random] versuchen, in [den item] zu steigen. Das geht z.B. mit STEIG IN ([den item])[roman type].";
		stop the action;
	if item is container:
		if else to say is true:
			say "[line break]";
			now else to say is false;
		say "[italic type]Du könntest [one of]vielleicht[or]mal[or]eventuell[or]einfach[or]jetzt[at random] versuchen, etwas in [den item] zu legen. Dafür gibt es den Befehl LEGE (etwas) IN ([den item])[roman type].";
		stop the action;			
	if item is edible:
		if else to say is true:
			say "[line break]";
			now else to say is false;
		say "[italic type]Du könntest [one of]vielleicht[or]mal[or]eventuell[or]einfach[or]jetzt[at random] versuchen, [den item] zu essen. Das geht z.B. mit ESS ([den item])[roman type].";
		stop the action;	
	if item is a person:
		if else to say is true:
			say "[line break]";
			now else to say is false;
		say "[italic type]Na prima, jetzt ist endlich eine weitere Person hier. Du könntest [one of]vielleicht[or]mal[or]eventuell[or]einfach[or]jetzt[at random] versuchen, mit [dem item] zu reden. Versuche SAG (etwas) ZU ([dem item]), REDE MIT ([der item]),oder ERZÄHLE ([dem item]) VON (etwas) oder FRAGE ([den item]) NACH (etwas)[roman type].";
		stop the action;
	if item is portable:
		if else to say is true:
			say "[line break]";
			now else to say is false;
		say "[italic type]Du könntest [one of]vielleicht[or]mal[or]eventuell[or]einfach[or]jetzt[at random] versuchen, [den item] mitzunehmen.  Das geht z.B. mit NIMM ([den item])[roman type].";
		stop the action;
	if item is fixed in place or item is scenery:
		say "[italic type]Betrachte [den item] genauer mit B ([den item])[roman type].";

Section - Asking for help and hinting

Asking for help is an action applying to nothing.

Understand "Info" or "Information" as asking for help.

Instead of asking for help:
	say "'Mariel' ist als Kurzgeschchte geschrieben worden, um ein paar Grundlagen von GerX, der deutschen Extension zu Inform7, zu erläutern und besteht aus ungefähr 10000 Wörtern Quellcode. Mehr über GerX und das Team erfahrt ihr unter http://ifiction.pageturner.de/inform7/.[paragraph break]Viel Spaß beim Lesen.".

Asking for credits is an action applying to nothing.

Understand "Credits"  or "Danke" or "Danksagung" as asking for credits.

Instead of asking for credits:
	say "Mariel SE verwendet folgende Erweiterungen für Inform 7:[paragraph break]- Flexible Windows by Jon Ingold[line break]- Glimmr Graphic Hyperlinks by Erik Temple[paragraph break]Ein Dankeschön für intensives Betatesting und viele gute Anregungen geht an Martin Barth, Christian Blümke, Michael Graff, Martin Oehm und Oliver Ullmann.[paragraph break]Der Autor dankt weiterhin allen, die an der Entwicklung und Verbreitung von Inform mitgewirkt haben - über allem gilt der Dank Graham Nelson."

Hinting is an action applying to nothing.

Understand "Hinweis"  or "Hinweise" or "Hint" or "Hints" or "Help" or "Hilfe" as hinting.

Instead of hinting for the first time:
	say "Wenn du nicht weiter weißt, kannst du zunächst DENK NACH eingeben. Dann erhältst du eine Liste der Dinge, die sich in der Nähe befinden. Möchtest du konkrete Hinweise zur Lösung erhalten, gib HILFE ein.".

Instead of hinting:
	if the player does not know about the nurse:
		say "Du solltest erst mal erkunden, was sich hinter der weißen Tür befindet.";
		stop the action;
	if the player can see the nurse:
		say "Hier hilft offensichtlich nur eins: sich wieder hinzulegen.";
		stop the action;
	if the window is open and the location is the hospital:
		say "Klettere doch einfach durch das Fenster.";
		stop the action;
	if the window is open and the location is west end:
		say "Geh zurück ins Zimmer und klettere doch einfach durch das Fenster.";
		stop the action;
	if the window is open and the location is ventilation system:
		say "Geh zurück ins Zimmer und klettere doch einfach durch das Fenster.";
		stop the action;
	if the crowbar is in the station-cabinet and location is in indoors:
		say "[The nurse] hat die Brechstange an einen sicheren Ort gebracht. Du kannst diesen Ort über den Lüftungsschacht erreichen[if hole is not in hospital]. Steig auf das Bett und berühre die Decke[end if].";
		stop the action;
	if the player can see the crowbar and location is in indoors:
		say "[if the player is holder of the crowbar]Brich das Fenster auf[otherwise]Nimm die Brechstange und brich das Fenster auf[end if].";
		stop the action;
	if the bed is moved and the location is the hospital:
		say "Du hast das Bett bereits gedreht. Das ist der erste Schritt in die Freiheit. Durchsuche das Zimmer nochmals genauer.";
		stop the action;
	if the bed is not moved and the location is hospital:
		say "Der 'Schlüssel' zur Freiheit ist in dem alten Bett versteckt. Das Bett hat eine bestimmte Ausrichtung. Versuche, diese zu ändern.";
		stop the action;
	if the player is not holder of the crowbar and location is in indoors:
		say "Hole die Brechstange und brich das Fenster.";
		stop the action;
	if location is in outdoors and the player does not know about the caretaker:
		say "Versuche, den Eingangshof zu erreichen.";
		stop the action;
	if location is in outdoors and the caretaker is not in the pit:
		say "Stelle dem Koloss eine Falle. Versuche die Falltür zu tarnen.";
		stop the action;
	if location is in outdoors:
		say "Verlasse diesen Ort so schnell wie möglich.";
		stop the action;
	if location is police car:
		say "Sei vorsichtig, bei allem was du tust und sagst. Bestimme Gesprächsthemen verschlimmern die Lage, andere lassen den Polizeibeamten einsichtig werden. Leg vor allem die Brechstange ab, bevor du den Polizeiwagen betrittst.".

Section - Atmosphere bonus

At 09:24 AM:
	if location is hospital:
		say "Du hörst seltsame Geräusche aus dem Nebenzimmer.".

At 09:25 AM:
	if location is hospital:
		say "Ein seltsamer, langgezogener, verzweifelter Schrei hallt durch den Korridor.".

At 09:26 AM:
	if location is hospital:
		say "Stimmengewirr. Was ist da draußen bloß los?".

At 09:27 AM:
	if location is hospital:
		say "Du fragst dich ernsthaft, ob du verrückt wirst. Ein Gedanke kommt dir plötzlich in den Sinn: was, wenn dies hier gar kein Krankenhaus ist? Nein, das kann nicht sein. Beruhige dich. Es ist sicher alles in Ordnung.".

At 09:28 AM:
	if location is hospital:
		say "Deine Nerven haben überreagiert; sicher eine Nachwirkung des Schocks, den zu zweifelsohne nach dem Unfall hattest.".

Instead of searching the window:
	say "Draußen ist alles schwarz; du kannst kaum etwas erkennen.".

Instead of jumping when location is hospital:
	say "Du springst in dieser engen Kammer ein wenig umher, um deine Muskeln zu lockern.".

Instead of jumping when location is in outdoors:
	say "Du springst hier draußen ein wenig umher, um deine Muskeln zu lockern.".

Instead of jumping when location is police car:
	say "Hier ist wirklich kein Platz, um zu springen.".

Instead of singing:
	say "Du summst ein Lied, um dir Mut zu machen.".

Instead of eating the window:
	say "Aus Verzweifllung nagst du am Gitter. Zwecklos.".

Instead of looking under the bed:
	say "Nichts als Staub.".

Instead of looking under the cabinet:
	say "Nichts als Staub, Staub, und nochmals Staub.".

Instead of waiting:
	say "Und wieder verstreichen wertvolle Minuten.".

Instead of waving when location is hospital:
	say "Du winkst aus dem Fenster. Niemand sieht es.".

Instead of waving when location is in outdoors:
	say "Du winkst in die Ferne. Niemand sieht es.".

Instead of waving when location is police car:
	say "Du winkst dem Polizisten zu. Seinem Blick zufolge hat das deine Glaubwürdigkeit nicht gerade erhöht.";
	increase suspiciousness of the policeman by 1.

Instead of pushing or pulling or attacking the white door:
	say "Du hämmerst wie wild auf die Tür ein. Zwecklos.".

Instead of pushing or pulling or attacking the policeman:
	say "Das wäre jetzt wirklich keine gute Idee.".

Instead of listening when location is hospital:
	say "Du hörst immer noch das beständige leise Piepen im Hintergrund. Es kommt irgendwie von draußen.".

Instead of listening when location is in outdoors:
	say "Du hörst nichts; kein Vogelgezwitscher, keine Autos in der Ferne, kein Heulen des Winds, nichts. Gespenstisch.".
	
Instead of listening when location is police car:
	say "Das Autoradio spielt leise 'Come home, Baby'.".

Instead of smelling when location is hospital:
	say "Ein leichter antiseptischer Geruch liegt in der Luft.".

Instead of smelling when location is in outdoors:
	say "Ein modriger Geruch liegt in der Luft.".

Instead of smelling when location is police car:
	say "Der Geruch von Zigarrenrauch liegt in der Luft.".

Instead of waking up:
	say "Wenn das ganze nur ein Traum wäre!".

Instead of exiting from the bed when the player is not on the bed:
	say "Du bist schon aufgestanden.".

Instead of exiting when the player is not on the bed and the player is not in the cabinet and location is hospital:
	if the window is open:
		try going north;
		stop the action;
	if the white door is open:
		try going south;
		stop the action;
	say "Wenn du bloß wüßtest, wie!".

After examining the window for the first time:
	say "Beim Betrachten dieses Fensters erinnerst du dich plötzlich dunkel daran, mehr Fenster in dieser Art gesehen zu haben. Du warst auf dem Rücksitz eines Autos festgeschnallt, das auf ein großes, schwarz gestrichenes Gebäude mit hunderten dieser Fenster zugefahren ist.".

After examining the white door for the first time:
	say "Plötzlich erinnerst du dich: du warst auf einer Liege festgeschnallt und wurdest durch endlose Korridore mit weißen Türen geschoben, bis sie dich schließlich in dieser Kammer abgelegt haben. Irgendeinen süßlichen Tee hat man dir zu trinken gegeben und danach sind alle Erinnerungen verschwunden.".

After examining the crowbar for the first time:
	say "Du erinnerst dich an letzte Nacht; an den strömenden Regen; an die seltsame Person, die dich mit einer solchen Brechstange aus deinem Auto befreit hat. Sie hat dich schließlich gepackt und auf einer Liege festgeschnallt. Ist das alles wirklich passiert?".

After examining the bed for the first time:
	say "Dieses Bett weckt unangenehme Erinnerungen; Erinnerungen an lange Korridore mit vielen Kammern wie dieser, leblose Gestalten lagen auf diesen Betten. Eins ist gewiss: du musst schnellstens hier raus.".

After opening the white door:
	say "Du öffnest leise die Tür.[paragraph break]Vorsichtig schaust du um die Ecke - niemand zu sehen.".
	
Instead of touching the nurse:
	say "Alleine der Gedanke ist grausig.".

Instead of touching the caretaker:
	say "Alleine der Gedanke ist grausig.".
	
Instead of touching the policeman:
	say "Du berührst den Polizeibeamten; nein, es ist kein Traum. Seinem Blick zufolge hat das deine Glaubwürdigkeit nicht gerade erhöht.";
	increase suspiciousness of the policeman by 1.
	
Instead of pushing or pulling or turning the policeman:
	say "Du rüttelst den Polizeibeamten an der Schulter; nein, es ist kein Traum. Seinem Blick zufolge hat das deine Glaubwürdigkeit nicht gerade erhöht.";
	increase suspiciousness of the policeman by 1.
	
Instead of kissing the nurse:
	say "Niemals.".

Instead of kissing the caretaker:
	say "Der Albtraum schlechthin.".

Instead of kissing the policeman:
	say "Aus Freude küsst du den Polizisten auf die Wange. Seinem Blick zufolge hat das deine Glaubwürdigkeit nicht gerade erhöht.";
	increase suspiciousness of the policeman by 1.

Instead of pushing or pulling or turning the nurse:
	say "Diese Frau sieht sehr muskulös aus. Du hättest keinerlei Chance, an ihr vorbei zu kommen.".

Instead of attacking the nurse:
	say "In letzter Verzweiflung stürzst du dich auf sie. Zu spät merkst du dass du ihr nicht gewachsen bist. Ihre starken Hände ringen dich zu Boden und der darauf folgende Schlag ins Gesicht läßt dich das Bewußtsein verlieren.";
	end the story finally saying "Du bist verloren.".

Instead of attacking or pushing or pulling the fence:
	say "Du rüttelst heftig an dem Zaun - vergeblich. Er ist solide gebaut.".

Instead of attacking:
	say "Das Letzte, was du jetzt brauchst, ist die Kontrolle über dich komplett zu verlieren.".

After reading a command:
	if the player's command includes "bitte":
		say "Bitte sag nicht immer bitte.";
		stop the action.
